
本文書に関して

Konohaをご試用いただきましてありがとうございます。
本書は、Konoha 言語のプログラミングガイド、及びその言語仕様をまとめたデザイ
ナーズノートです。長らく、Konohaにはマニュアルがなく、ご不便をかけてきましたが、
本書が Konohaの体験を広げる一助になれば幸いでございます。なお、本文書は、「書き
かけでもいい」から少しでも多くの情報を提供してほしいという要望に応え、本当に「書

きかけ」だったりしてお見苦しいところがたくさんあります。あらかじめ、お詫びしてお

きます。

また、Konoha言語自体、なおも開発が行われています。本書は、Konoha 1.0仕様を前
提に書いており、仕様が先行しているため、実装との間にギャップがあります。そこで、

各節には*, **のようにマークを付けてあり、それぞれ部分的な開発/テストが残っている
機能、計画されているが実装されていない機能を表しています。

Konohaユーザメーリングリストでは、Konohaに関する最新情報を扱っております。是
非、ご登録の上、本書の不備や最新機能に関する情報を得て下さい。また、バグレポート

も歓迎しております。

アドレス: konoha-users@sourceforge.jp
登録: http://lists.sourceforge.jp/mailman/listinfo/konoha-users

著作権表記

Copyright (c) 2008-2009 Kimio Kuramitsu, Konoha Project. All rights reserved.

再配布の留意事項

本書は、現在 PDF 版を Konoha プロジェクトの sourceforge サイトから配布していま
す。将来、出版してくれる出版社が見つかれば、出版したいと考えていますが、電子版の

配布もユーザ数の動向を把握するため、可能な限り継続する予定です。ご知人に紹介する

ときは、PDFファイルを直接、再配布する代わりに、一次配布サイトからダウンロードす
るように薦めて頂けますようにお願いします。



1

http://sourceforge.jp/projects/konoha/releases/





3

目次

第 1章 はじめに 1
1.1 Konohaの始まり* . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 1

第 2章 Konohaの特徴 3
2.1 おなじみの言語文法 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 3
2.2 対話的な動作: Looks like Java, Runs like Python . . . . . . . . . . . . . . 4
2.3 型推論 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 5
2.4 最速水準の実行性能 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 6
2.5 高いポータビリティ* . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 7
2.6 実行前の型検査といつでも実行 . . . . . . . . . . . . . . . . . . . . . . . 7
2.7 オブジェクト指向 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 8
2.8 活版フォーマッタ* . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 9
2.9 マッピング機能とデータ変換* . . . . . . . . . . . . . . . . . . . . . . . . 10
2.10 セマンティック・プログラミング* . . . . . . . . . . . . . . . . . . . . . . 10

第 3章 レキシカル構造 11
3.1 文字 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 11
3.2 トークン . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 11
3.3 ステートメント . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 12
3.4 コメント . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 13
3.5 リテラル . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 13
3.6 識別子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 14
3.7 予約語 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 18

第 4章 オペレータ 19
4.1 代入演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 19
4.2 コール演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 20
4.3 アドレス演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 21
4.4 比較演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 21
4.5 論理演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 23



4 目次

4.6 算術演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 24
4.7 ビット演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 26
4.8 シーケンス演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 27
4.9 スライシング演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 28
4.10 キャスト演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 28
4.11 イテレータ演算子* . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 29
4.12 セマンティック演算子** . . . . . . . . . . . . . . . . . . . . . . . . . . . 30

第 5章 ステートメント 31
5.1 式によるステートメント . . . . . . . . . . . . . . . . . . . . . . . . . . . 31
5.2 ブロック . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 32
5.3 変数の型宣言 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 32
5.4 if/else文 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 34
5.5 switch文** . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
5.6 while/do-while文 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 37
5.7 for文 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 37
5.8 break文と continue文 . . . . . . . . . . . . . . . . . . . . . . . . . . . 38
5.9 for/in文 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 39

第 6章 数値 41
6.1 数値と型 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 41
6.2 数値リテラル . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 42
6.3 数値演算 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 44
6.4 数値フォーマッタ . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 48
6.5 乱数生成 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 50

第 7章 文字列 51
7.1 文字列と型 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 51
7.2 文字列リテラル . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 52
7.3 文字列と演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 56
7.4 文字列メソッド . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 58
7.5 テンプレート・フォーマッティング . . . . . . . . . . . . . . . . . . . . . 61
7.6 インライン・フォーマッティング . . . . . . . . . . . . . . . . . . . . . . 63
7.7 正規表現 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 63

第 8章 配列とリスト 67
8.1 配列と型 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 67
8.2 配列の生成 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 69
8.3 配列と演算子 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 70
8.4 多次元配列 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 71



5

8.5 配列とメソッド . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 73
8.6 バイト配列: byte[] . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 75

第 9章 関数 77
9.1 関数の定義 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 77
9.2 関数コール . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 80
9.3 ビルトイン関数 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 80

第 10章 クラスとオブジェクト 85
10.1 class宣言 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 85
10.2 フィールド変数 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 86
10.3 メソッド . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 89
10.4 クラス関数 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 92
10.5 コンストラクタ . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 93
10.6 オペレータとメソッド . . . . . . . . . . . . . . . . . . . . . . . . . . . . 93

第 11章 クラス階層と抽象化 95

第 12章 例外処理 97
12.1 例外クラス Exception . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 97
12.2 throw文 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 97
12.3 try-catch文 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 98
12.4 finally節 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 99

第 13章 スクリプトと名前空間 101
13.1 名前空間 NameSpace . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 101
13.2 スクリプトクラス Script . . . . . . . . . . . . . . . . . . . . . . . . . . . 103
13.3 スクリプト変数 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 103

第 14章 デバッグ 107
14.1 DEBUGブロックとデバッグモード . . . . . . . . . . . . . . . . . . . . . 107
14.2 print文 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 109
14.3 assert文 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 110
14.4 utest文** . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 112
14.5 ブレークポイント** . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 112

第 15章 Konohaライブラリ 113
15.1 Konohaインスタンス . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 113
15.2 Cからのスクリプト関数の利用* . . . . . . . . . . . . . . . . . . . . . . . 115

第 16章 C/C++ライブラリの利用 117



6 目次

16.1 C言語関数のバインド . . . . . . . . . . . . . . . . . . . . . . . . . . . . 117
16.2 グルー関数とメソッド . . . . . . . . . . . . . . . . . . . . . . . . . . . . 119
16.3 構造体とクラス . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 119
16.4 関数ポインタとクロージャ . . . . . . . . . . . . . . . . . . . . . . . . . . 119

参考文献 121

付録 A 入手方法 123

索引 124



1

第 1章

はじめに

Konohaプロジェクトはどのように始まったのか？その背景にある言語設計のフィロソ
フィーに関して説明する。

1.1 Konohaの始まり*

Konohaの開発は、ある体験から始まった。（以下続く）

第 2章

Konohaの特徴

Konoha の特徴は、ひと言で言えば、「静的に型付けされたスクリプティング言語」で
ある。しかし、スクリプティング言語は、別名「ダイナミック言語」と呼ばれるとおり、

動的に型付けされている。そのため、Konohaは、どこかしら保守的でもあり、逆に新し
さもあるスクリプティング言語となっている。本章では、Konohaの特徴を簡単なソース
コードとともに紹介したいと思う。

2.1 おなじみの言語文法

Konohaの文法は、C/C++や Javaと高い互換性をもっている。これは、はじめてプログ
ラミングを学ぶ者にとって、C/C++, Javaなどのメインストリーム言語の良き導入になる
だけでなく、経験あるプログラマにとって特別な努力を費やすことなく、Konohaプログ
ラミングがはじめられることを意味する。

次は、おなじみ fibonacci数列を再帰で求める関数である。C/C++, Javaでも、もちろん
Konohaでも全く同じように実行することができる。

int fibo(int n) {
if(n < 3) return 1;
return fibo(n-1)+fibo(n-2);

}

一方、これが、Perl や Python, Ruby, Lua など、他のスクリプティング言語であった
ら、やはり独特な文法のため、いきなり簡単な fibonacci関数も書くことは難しい。次は、
「Java風の文法をもった」といわれる、JavaScriptによる fibonacci関数の例であるが、は
やり無視できない大きな違いがいくつもある。

function fibo(n) { // JavaScript の場合
if(n < 3) return 1;

return fibo(n-1)+fibo(n-2);
}

Konohaが、従来の Java風スクリプティング言語と比べて、高い文法互換性をもってい
る理由は、Javaと同じく静的に型付けされている点に大きい。これによって、オブジェク
ト指向プログラミングにおいても、より Javaに近いモデリング手法でクラスを設計する
ことができる。

次は、Counter クラスの定義の例である。クラス宣言に始まり、コンストラクタから
フィールドまで、通常のプログラミングにおいて、まず問題にならないレベルの互換性が

ある。

class Counter {
int cnt;
Counter(int n) { cnt = n; }
void count() { cnt++; }
void reset() { cnt = 0; }

}

ただし、Konohaは、独立したプログラミング言語である。文法の互換性は、あくまで
も C/C++や Javaプログラマへの混乱を最小限にとどめるために導入されている。もちろ
ん、スクリプティング言語として、簡略化されている部分もあるし、独自に拡張されてい

る文法もある。

2.2 対話的な動作: Looks like Java, Runs like Python

Konohaプロジェクトのキックオフ当時の目標は、”Looks like Java, Runs like Python”、
つまり「Javaのように書けて Pythonのように実行できる」であった。現在、Konohaは、
Pythonのみならず、Rubyや JavaScriptなど、様々な言語の「よい設計」を取り込んでい
るが、Konohaのスクリプティング言語としてのお師匠といえば、Pythonなのである。

Pythonへのオマージュが最もよくあらわれている部分が、対話モードである。konoha
コマンドを実行すると、pythonコマンドと同様に対話モードが起動する。

$ konoha
Konoha 0.3.10 (Rufy) GPL2 (#544, May 2 2009, 09:37:22)
[GCC 4.0.1 (Apple Inc. build 5490)] on macosx_32 (32, UTF-8)
Options: iconv sqlite3 thread regex used_memory:412 kb
>>>

>>>は、コマンドプロンプトである。複数行にわたる場合は、2行目以降は、...とな
る。ここにプログラム（式もしくはステートメント）を入力すると、次の行にはその実行

結果が得られる。

>>> print "hello, world"
hello, world

Konohaの対話モードは、Lispの対話的プログラミングほど、それ自体で成り立つプロ
グラミング環境ではない。しかし、Konohaの機能、動作を試したいとき、十分に役立つ
機能である。もちろん、先ほどの fibonacci関数も、対話モードで定義し、その場で実行
することもできる。

>>> int fibo(int n) {
... if(n < 3) return 1;
... return fibo(n-1)+fibo(n-2);
... }
>>> fibo(10)
55

2.3 型推論

Konohaの特徴は、静的な型付けである。これは、ごく一般的に言えば、変数の型宣言
をプログラマに強制することを意味する。ダイナミック言語に親しんできたプログラマに

は、これは何かしら面倒な話に聞こえてしまうだろう。

int n = 1; // 変数宣言

念のため補足しておけば、静的な型付けは、プログラミングモジュールの仕様を明示的

に定義することであり、大規模なソフトウェア開発やグループ開発において、ケアレス

ミスや誤解によるソフトウェアクラッシュを防ぐ重要な機能である。Konohaがスクリプ
ティング言語であっても、静的に型付けされているのは悪い話ではない。

それでも、慣習とは恐ろしいもので、次のようにいきなり変数を使いたくなることも

ある。

n = 1; // 変数宣言なし

Konohaは、非常に限定的であるが、実用的に十分な型推論 (type inferencing)の機構を
備え、明示的に変数宣言をしなくても、その変数の型を推論する機能を備えている。これ

は、変数宣言なしであっても、変数の初期化の値からその変数の型を推論する機能である。

>>> a = [1,2,3] // 推論による型宣言
>>> n = a[1]
>>> typeof(a) // 静的な型付けを調べる
Int[]
>>> typeof(n)

Int

Konohaの型推論は、プログラミングのしやすさと型宣言による厳密な仕様定義の間で、
バランスをとりながら設計されている。また、型を推論できなかった場合、Any型による
「動的な型付けのオプション」も用意されている。

>>> s = null; // 型推論できない
>>> typeof(s)
Any

Any型やダイナミック言語の特性を活かしたプログラミングは、「第 ??ダイナミック・
スタイル」でより詳しく解説する。

2.4 最速水準の実行性能

スクリプティング言語は、元来、プログラムの実行速度より、プログラムの開発速度へ

価値をおいた言語設計であった。そのため、実行性能を単純に比較すると、C/C++や Java
の足下にも及ばない。しかし、今日、より複雑なアプリケーション開発にもスクリプティ

ング言語が利用されるようになるにつれ、スクリプティング言語エンジンの性能向上は大

きな関心事になっている。

Konohaは、バイトコード型バーチャルマシンで実装されたスクリプティング言語エン
ジンをもっている。スクリプトは、バイトコードにコンパイルされ、そしてバーチャルマ

シン上で実行される。

次は、コンパイル済みのバイトコード (fibo()関数)をフォーマッタ機能でダンプし
た様子である。

>>> %dump(Script.fibo)
int main.Script.fibo(int n)
[4:0x2038e0] SETESP(30) sfp[2]
[4:0x2038e8] iLTn(99) sfp[2] sfp[1] 3
[4:0x2038fc] bJIFF(62) 0x20391c sfp[2]
[4:0x203910] RETo(35) sfp[-1] 1
[5:0x20391c] iSUBn(87) sfp[5] sfp[1] 1
[5:0x203930] FCALL(46) sfp[3] 3 sfp[0] int fibo(int n)
[5:0x203940] iSUBn(87) sfp[6] sfp[1] 2
[5:0x203954] FCALL(46) sfp[4] 3 sfp[0] int fibo(int n)
[5:0x203964] iADD(84) sfp[-1] sfp[3] sfp[4]
[5:0x203970] RET(32)

Konohaの大きな特徴は、静的な型付けの特性を活かして、ダイナミック言語より効率
のよい実行コードを生成している点である。10年以上にわたる高速化チューニングが施
された Python や世界最速のスクリプティング言語 Lua と比べも、図 2.1 に示すとおり、
数倍以上の実行性能を示している。

Konohaが大学の研究室生まれであり、プロフェッショナルな開発者が参加していない
現状を勘案すれば、より大幅なチューニングも可能と予想される。また、スクリプティン

図 2.1 fibo(36) による性能比較: fibonacci 数列は、数値演算だけでなく、ローカル変
数へのアクセスや関数コールなど、プログラミング言語の基礎的な性能を図ることに広

く利用されている。

グ言語の高速化に関しては、学術的なアプローチにより、新しい高速化技法の研究も進め

られている。

2.5 高いポータビリティ*

Konoha プロジェクトは、開発者が、TRON プロジェクト出身であったこともあり、
TRON OS を含めた組み込み分野やユビキタス応用を視野にいれた言語エンジンの開発
を行ってきた。そのため、Konohaは、ポータビリティの高い Cソースコードで書かれ、
UNIX系の OSだけでなく、Windowsや TRON(T-Kernel)などで、様々なプラットホーム
で動作が確認されてきた。(図 refteaboard)

図 2.2 パーソナルメディア製 Teabord (OS:T-Kernel, CPU:ARM, Memory:16Mb)の上
で動作する Konohaスクリプティング言語エンジン (2008年頃撮影)

labelteaboard

2.6 実行前の型検査といつでも実行

型検査とは、型に対する操作、例えば関数コールのとき引数の数や種類が間違っていな

いか確認することである。ダイナミック言語は、実行時にのみ型検査が行われるため、型

エラーが含まれるスクリプトでも実行しないとエラーが発見できない。そのため、全ての

実行パスをひとつひとつテスト実行しながら、型エラーを探し、修正する必要がある。

Konoha は、静的な型付け言語であるため、スクリプトの実行前に型検査を機械的に、
つまりコンパイラが行うことができる。また、スクリプトを実行することなしに、型エ

ラーや潜在的なミスを全て検証するための専用のオプション (-c)も用意されている。こ
のとき、実際にプログラムが走り始めることはないため、ファイル操作やデータベースが

中途半端な状態でプログラムが停止する弊害もない。

$ konoha -c sample/err.k
konoha -c err.k
- [err.k:4]:(errata) added return value
- [err.k:7]:(error) type error: Int is not string

Konoha は、従来の静的な型付け言語と異なる点もある。”Run anytime”コンパイラ技
術を採用し、コンパイル中に型エラーが発見されても、型エラーの部分のみ、実行しても

安全なコード（ランタイム例外）に書き換えることで、いつでも実行可能なバイトコード

を生成することができる。そのため、コンパイルが通らないため実行できないということ

はなく、スクリプティング言語の開発しやすさが保たれている。

Konohaでは、もしエラー箇所を実行したときは、Source!!例外が通知されて停止す
る。プログラマは、エラーを（全部）取り除いてから実行するか、それとも試しに実行さ

せながらバグをとるか、どちらのスタイルの開発でも選ぶことができる。

$ konoha sample/err.k
- [err.k:7]:(error) type error: Int is not string
** Source!!: Running errors at [err.k:7]

2.7 オブジェクト指向

Konohaは、「全てがオブジェクト」という世界観で統一されたオブジェクト指向プログ
ラミング言語である。整数も nullもオブジェクトで表現されている。また、クラスルーム
利用において、最先端のオブジェクト指向プログラミングのアイディアを学ぶことができ

るように設計されている。

• 名前ベースのクラスシステム
• 単一継承、インターフェースによるポリモーフィズム
• ダックタイピングによるポリモーフィズム
• 総称型（Generics)
• メタオブジェクト、アスペクト指向
• データ変換と相互運用性（Mappable Class)



2.8 活版フォーマッタ* 9

2.7.1 クラスを調べる

オブジェクト指向プログラミング言語は、各クラスにその機能性が集約されている。

Konohaでは、対話モードからmanコマンドで使えば、クラスが提供するオペレーション
が表示される。

>>> man Int
CLASSNAME

konoha.Int
extends konoha.Number

CONST
Int.MAX: 9223372036854775807
Int.MIN: -9223372036854775808

OPERATOR
-x x != x x & y x * y x + y

x - y x / y x < y x << y x <= y
x == y x > y x >= y x >> y x ˆ y

x mod y x | y x++ x-- |x|
˜x

METHOD
Int! Int.random(Int n)

FORMATTER
%bits %d %f %s %u %x

MAPPING
Int

==> Float
==> String

Konohaは、ソフトウェアモジュールの柔軟性のため、実行時にメソッド等を追加でき
る。manコマンドは、実行時の状態を表示するため、メソッドの種類は変更になる。

2.8 活版フォーマッタ*

活版印刷 (movable type) は、出版の普及による中世から近代への橋渡しの役割を果た
したため、コンピュータ発明以前の最大の情報技術イノベーションと考えられている。

Konohaでは、HTMLや XML文書の生成のため広く利用されるテキストフォーマッティ
ングにおいて、活字と同様に再利用性の高い手法を提供することを目指して、組み合わ

せが自由自在な「活版フォーマッティング (movable formating)」技術を新たに導入して
いる。

基本的なアイディアは、オブジェクト指向モデルに基づく「書式付きフォーマッタ」に

ある。フォーマッタは、C言語の printf風の書式 (%s, %d)に似ているが、オブジェクト
と書式は出力時にダイナミックバインディングされる。つまり、C言語では型に対して書
式が固定的に決まっていたのに対し、Konoha では int型でも小数形式で出力したいとき
は %fを用いることができる。

>>> %s(1)
1
>>> %03d(1)
001

>>> %f(1)
1.00000
>>> %bits(1)
00000000 00000000 00000000 00000001

オブジェクト単位のフォーマッタを組み合わせて、自由自在にフォーマッティングを行

うことができる。

>>> p = 9.80
>>> ‘<price> $%.2f{p}</price>‘
"<price>$9.80</price>"

また、フォーマッタはメソッドの一種として、プログラマが自由に定義することもでき

る。%XMLのような複雑なフォーマッタも作成できる。

>>> format %XML (Catalog c) """
<product>

<title> %s{c.name}</title>
<price> $%.2f{c.price}</price>

</product>
"""

>>> %XML(c)
<product>
<title>Harry Potter DVD</title>
<price>$9.80</price>

</product>

注意：フォーマッタは、ストリームへの出力という形でフォーマッティングを行うため、

大量のデータを低メモリ消費で変換することができる。詳しくは、「第 ??章活版フォー
マッタ」で述べる。

2.9 マッピング機能とデータ変換*

2.10 セマンティック・プログラミング*

第 3章

レキシカル構造

3.1 文字

Konoha は、はじめから多国語環境での利用を前提として設計されている。文字は、
Unicode規格で定められた文字コードが利用可能であり、スクリプトの記述には Web上
で標準である UTF-8エンコーディングを採用している。また、Konohaの実行環境は、文
字コード変換 (iconvもしくは相当)機構と統合されており、オペレーティングシステムの
ローカルなエンコーディングも自動的に UTF8に変換可能して入出力することができる。
さて、Konohaは多国語前提と言っておきながら、識別子（変数名、クラス名、メソッド
名）には ASCII 文字の利用しか認めていない。これについてちょっと説明を加えておき
たい。そもそも、Java言語が登場したとき、識別子に Unicode文字、つまり「日本語名」
を認めた。これは、たぶん、彼らなりの非欧文圏への親切だったと思うのあるが、それか

ら 10年以上を過ぎた現在、日本語で書かれた変数やクラス名をみる機会は全くない。こ
こから学ぶ教訓は、欧文圏のプログラマにとって「入力すらできない名前」を使わないこ

とも、我々側からの親切であるといえる。

3.2 トークン

トークンとは、プログラミング言語の最小の意味単位である。それぞれ、予約語（Konoha
文法で定義されたキーワード）、識別子、演算子、リテラル（データ値）などの意味を持

つ。ソースコードは、Konohaの文法ルールにしたがって、トークンに分割される。

return 0; // 3つのトークン "return" "0" ";"
return0; // ２つのトークン "return0" ";"

空白は、ソースコードからトークンを分割するための区切りとなる記号で、空白はいく

つ続いても意味はない。タブ (Tab)や改行も空白と同様に扱われる。そのため、プログラ

マは自由に空白、タブ、改行を用いて読みやすいソースコードを書くことができる。ただ

し、次節で述べるとおり、改行に関してはひとつだけ留意事項がある。

また、次の文字は前後に空白があるものとみなされる。

{ } ( ) [ ] ; ,

また、演算子で用いられる記号 (+や*など) の前後にも空白があるものと解釈される。
ただし、読みやすいソースコードを書くため、明示的に空白をいれることが推奨される。

a + b // 推奨
a+b // 非推奨

3.3 ステートメント

ステートメントとは、プログラミング言語における文のことである。Konohaのステー
トメントは、Cや C++, Javaと同様に、セミコロン (;)で終わる。

a = 3;
b = 4;

次のように、１行に複数のステートメントを書くこともできる。

a = 3; b = 4;

ここまでは、Cや C++, Javaで採用された正しいスタイルのステートメントである。も
し、Konohaを学習用として利用している場合、100%この正しいスタイルを身につける
べきである。

3.3.1 セミコロンの省略

多くのスクリプティング言語は、1 行の終わりをそのままステートメントの終わりと
解釈することが多い。そこでセミコロンを強制することは、Konohaは書きにくい言語と
なってしまう。そこで、Konohaは、ステートメントの途中に改行がくると、セミコロン
(;)を忘れたものと拡大解釈している。つまり、次の例でも、同様に、２つのステートメン
トになる。

a = 3
b = 4

こうすると、あちらを立てればこちらが立たずという問題が発生する。

Cプログラマは、１行 (80文字)に収まりきらないステートメントを改行して読みやす
くする習慣があり、このケースではステートメントが勝手に分割されてしまうことになり

かねない。

print a + b
-c;

そこで、もしステートメントの途中で改行を入れたときは、２行目以降は字下げしてス

テートメントが続いていることを明示的に示すことにした。少々面倒な気もするが、多く

の Cプログラマは読みやすくするため、習慣的に字下げを行っているため、比較的影響が
少ないと思われる妥協点である。

print a + b
- c;

3.4 コメント

Konohaは、C/C++, Javaスタイルのコメントを採用している。//は、行コメントの始ま
りであり、改行までのテキストがコメントとして無視される。

a = b; // 行コメント

また、C/C++や Javaと同様に、/* */で囲むことでコメント化できる。注意すべき点
は、Konohaは C言語と異なり、コメントのネストが可能な点である。

/* まとめてコメントアウト
a = 1;
b = 2; /* コメントの入れ子も OK */

*/

3.5 リテラル

リテラルは、プログラム中に直接あらわれたデータ値のことである。型をもった値とし

て扱われる。

次は、基本的なリテラルの例である。

12 // 整数 12
1.2 // 小数 1.2
"hello world" // テキスト
/ˆworld$/ // 正規表現パターン
true // 論理値 true
false // 論理値 false
null // NULL

スクリプティング言語の一般的な特徴は、リテラルが一般的なプログラミング言語に比

べ大幅に強化されている点である。Konohaも、データはスクリプトの重要な一部とみな
し、配列 (Array),辞書 (DictMap)、そしてオブジェクトのリテラルをサポートしている。

[1,2,3] // 配列オブジェクト
{x :1, y: 2} // 辞書オブジェクト
C{x: 1, y: 2} // クラス Cのオブジェクト

リテラルは、組み合わせてより複雑なデータを記述することもできる。Konohaのデー
タリテラルは、JSONなどと同様にWeb上でのデータ変換を想定しているため、セキュリ
ティ上の理由から変数や式を含めることはできない。

3.6 識別子

識別子は、名前のことである。Konohaでは、識別子は、クラス、変数、メソッド、ア
ノテーションに名づけるために利用される。名前の最初は、アルファベット文字でで始ま

り、英数字かアンダースコア (_)が続く。次は、全て正しい識別子の例である。

i
my_variable_name
v9
Class
MATH_PI

Konohaの識別子の特徴は、Javaプログラミングの名前付けの慣習 (naming convension)
をルール化し、識別子からその種類を判定できるようにした点である。ただし、識別子の

一部はコンテキストから判断するしかない場合がある。

3.6.1 変数

変数は、常に英小文字で始まり、英数字、アンダースコアからなる名前をもつ。次は、

すべて正しい変数名の例である。

n n2 name firstName first_name

変数名の先頭にアンダースコアをつけると、それぞれフィールド変数やスクリプト変数

をダイレクトに参照する意味となる。

_name // フィールド変数
__name // スクリプト変数

3.6.2 定数

定数は、最初に与えられた値が変更されない特別な変数である。英大文字で始まり、英

大文字と数字、アンダースコアからなる。次は、正しい定数名の例である。

N N2 NAME FIRST_NAME

クラス定数は、クラス名とあわせて名付けられる定数である。次は、クラス定数の例で

ある。

Int.MAX Math.PI

3.6.3 クラス名

クラス名は、常に英大文字で始まり、英数字からなる。アンダースコア (_)を含めるこ
とはできない。次は、全て正しいクラス名である。

Int String InputStream C URN　

クラス名は、ローカル定数 (アンダースコアなしの定数) と、名前だけから区別するこ
とはできない。Konohaコンパイラは、コンテキストによって正しく判断することができ
るが、ソースコードの読みやすさを考慮に入れたとき、英大文字のみのクラス名 (C や
URN)の利用は避けた方がよい。

3.6.4 例外名

例外名は、!!で終わるクラス名である。次は、全て正しい例外名である。

Null!! Security!! OutOfIndex!! IO!!

Konohaは、Javaプログラマへの互換性を配慮して、クラス名が Exceptionで終わって
いた場合、パーサーが自動的に!! に置き換える。つまり、SecurityException は、

Security!!となる。逆に、Exceptionで終わる名前は、クラス名として用いること

はできない。

3.6.5 メソッド名 (関数名)

メソッド名*1は、変数と同様に、英小文字で始まる英数字からなる名前を用いる。

Konohaは、Javaスタイルの (英語の)動詞 +名詞でメソッド名を名付けるスタイルを採
用し、名詞の始まりで英大文字を用いることを推奨している。ただし、こればかりはコン

パイラでチェックできないので、プログラマの実践に任せるのみである。

次は、推奨スタイルに従ったメソッド名である。

get getName readLine query

メソッド名の途中でアンダースコアを入れることも認められている。ただし、挿入され

たアンダースコアは、取り除かれて、続く 1文字を英大文字化することで正規化される。

get_name // getName に正規化
get_host_by_name 　 // getHostByName に正規化
read_line // readLine に正規化

Konohaは、C/C++など他言語で開発されたライブラリをクラスにバインドして利用す
ることが少なくない。この場合、必ずしも Java風の名前慣習に従わないライブラリが多
く、既存のライブラリ関数をそのまま利用できた方が便利でもある。そこで、Konohaで
は、メソッドや関数をコールするときのみ、英大文字と英小文字の区別をしない。

次のどの関数呼び出しも、結局、getPid()を呼び出すことになる。

getpid()
GetPid()
GET_PID()

3.6.6 フォーマッタ名

フォーマッタ名は、%記号で始まる特別なメソッド名である。命名則は、メソッド名に

したがい、利用するときは英大文字/英小文字の区別はない。次は、フォーマッタ名の例で

*1 関数は、Scriptクラスのメソッドであるため、メソッド名のルールに従う。

ある。

%s %4.2f %dump %HTML

フォーマッタでは、利用のときに限り、%と名前の間に数字とドット記号 (.)から構成
される書式オプションを追加することができる。これは、C言語の printf書式の慣例にし
たがったものである。

3.6.7 プロパティ変数

プロパティ変数は、実行環境の設定値や環境変数をもった特別な変数名である。$記号

で始まり、英数字及びドット記号 (.)から構成される。次は、正しいプロパティ変数名で
ある。

$konoha.version
$env.PATH

3.6.8 アノテーション

アノテーションは、ステートメントに対して注釈やメタデータを与えるときに用いる。

アノテーション名は、Java言語と同様に、@で始まり、英数字からなる名前である。次は、
アノテーション名の例である。

@Override @Final @Doc

3.6.9 ラベル名

英字で始まりコロン (:)で終わるトークンは、予約語を含め、クラス名、変数名、定数
名とは関係なく、ラベルとして扱われる。次は、ラベルの例である。

default: ClassName: variable: CONST:

コロン (:)の前に空白を入れると、そのトークンはラベルと解釈されない。また、より
重要な注意は、ラベルの次は、必ず空白を入れることである。コロン (:)の次に空白がな
い場合は、連続したトークンとみなされる。

name : value // name はラベルでない

name: value // name: ラベル
name:value // ひとつの識別子

3.7 予約語

Konohaは、「識別子としてプログラム中で利用できない」特別なキーワードをいくつか
予約している。次のリストは、Konohaにとって文法の一部、つまり特別な意味をもった
キーワードである。

as assert break case catch class continue default defined do double else extends
false finally for foreach from goto if import include interface is isa lock namespace
new null pragma print return switch this true try typeof using var void where while

いくつかのキーワードは、Cや Java, JavaScriptなどの既存プログラミング言語の記法
を解釈するためのエイリアス（別名）として予約されている。

abstract // @Abstract の別名
boolean // Boolean
byte[] // Bytes
double // Float
final // @Final
float // Float
int // Int
private // @Private
protected // @Protected
public // @Public

第 4章

オペレータ

オペレータとは、1つ以上の式を結びつけて評価する機構である。+や=などの演算子
と呼び直した方がわかりやすいだろう。近代的なオブジェクト指向プログラミング言語

は、ほとんどのオペレータはメソッドの一部として実現し、それをシンタックスシュガー

（糖衣構文）として使いやすく提供している。本章では、Konohaが提供するオペレータの
全体像を知るため、一般的な紹介を行う。オペレータは、ある特定のクラスや機能と密接

に関係したものも多いが、それらは後に続く章において個々に詳説される。

4.1 代入演算子

代入演算子は、変数 xに右辺式 yを評価した結果を代入 (格納)する演算子である。

x = y // y の評価値を xに代入する

左辺値 xとなりうる値は、変数、プロパティ変数、フィールド、配列や辞書に限られる。

定数は、未定義の場合に限り、定数値を定義するために代入することができる。次は、代

入可能な右辺値の例である。

name = "naruto"
$name = "naruto"
p.name = "naruto"
p[0] = "naruto"
p["name"] = "naruto"
P = "naruto" // 定数は、1回のみ代入可能

4.1.1 代入式の代入*

Konohaでは、代入演算自体も式として扱うことができる。つまり、代入 x = y の評

価値は、(代入後の) xとなる。したがって、次のように、同じ値を複数の変数へ代入する
ことも可能となる。

>>> y = 2
>>> x = y = 1 // x には、(y = 1) の値が代入
>>> x
1
>>> y
1

4.1.2 多値セレクタ**

データ構造から部分的な値を切り出すことは多い。これらのプログラムを簡単に書くた

め、多値代入を値のセレクタとして利用する拡張計画がある。

>>> s = ["naruto", "sakura"]
>>> a, b = s;
>>> a // s[0] と同じ (s.opFirst())
"naruto"
>>> b // s[1] と同じ (s.opSecond())
"sakura"

4.2 コール演算子

コール演算子 ()は、メソッド/関数 f、コンストラクタ new C、もしくはフォーマッ

タ %fに続いて用いられ、メソッド/関数を呼び出すときに用いる。括弧 ()の中には、0
個以上のパラメータ/引数を,で区切って与えることができる。コール演算子の評価値は、
メソッド/関数の戻り値となる。(ただし、メソッド/関数の戻り値 void 型の場合は、評価
値はない。)

f() // 関数 f()のコール
o.f() // o のメソッド f()をコール
new C() // クラス Cのコンストラクタ
%f(o) // o のフォーマッティング

コール演算子で与えられるパラメータは、メソッド/関数ごとに個別に定義されている。
定義と異なるパラメータでコールしようとしたときは、型エラーとなる。

>>> fibo(10)
55
>>> Math.abs(-1)
1.000000

4.2.1 コール演算子の省略

コール演算子は、パラメータの数が１個であり、かつそれが文字リテラルである場合の

み、括弧 ()自体を省略することもできる。次は、c.query("""...""")と同じであ

るが、省略した方がすっきりとする。

>>> c.query """
select name, salary from PERSON_TBL
where age > 45 and age < 65;

"""

4.3 アドレス演算子

アドレス演算子は、変数 xの値が格納されているアドレスを返す。ただし、Konohaは
ポインタ演算をサポートしていないため、この機能は完全にクラスルーム目的で導入され

ている。

&x // x のアドレス

Konohaは、ポインタ型をサポートしていないため、アドレス演算子の演算結果は、int
型である。%pフォーマッタを使えば、アドレスとして表示される。

>>> s = "hello, world"
>>> %p(&s)
00000000

4.4 比較演算子

比較演算子は、二つの値 x, yを比較するとき用いられ、評価結果は、論理値の trueか
falseである。

x == y // 等しいか？
x != y // 等しくないか？
x < y // より小さいか？

x <= y // 以下か？
x > y // より大きいか？
x >= y // 以上か？

Konohaでは、全ての型において、x, yの比較が可能である。これは常に意味のある比
較となることを保証しない。特に、xと yの型が異なる場合、かつ概念的にも比較が意味
をなさない場合は、オブジェクト ID（アドレス値）の比較が返される。どちらにしても、
比較演算の評価値は、一意となる。

>>> 1 < 2.0 // 整数と浮動小数点は比較可能
true
>>> 1 == "1" // 比較不可能な型
false

4.4.1 マッチング演算子

マッチング演算子は、Perlに由来し、主に正規表現と文字列のマッチングに用いられる。

x =˜ y // xは、パターン yと等しいか？

マッチング演算子は、==と異なり、対称ではない。パターンは、必ず右辺に与える必要

がある。

>>> s = "hello, world"
>>> s =˜ /w.*d/
true

4.4.2 instanceof 演算子*

instanceof演算子は、Javaに由来し、xがクラス Cのインスタンスであるか判定す
る。Konohaは、純オブジェクト指向言語であり、全ての値は何らかのクラスのインスタ
ンスである。そのため、任意の xに対して、instanceofを用いることができる。

x instaceof T // x は、Tのインスタンスか？

整数 1 は、Float のインスタンスではないが、Number のインスタンスである。
instanceof演算子は、次のようにテストできる。

>>> 1 instanceof Float
false

>>> 1 instanceof Number
true

また、nullはどのクラスのインスタンスでないため、常に instanceof演算子の結果

は、falseとなる。ただし、Any型のみ trueとなる。

>>> null instanceof Object
false
>>> null instanceof Any
true

4.5 論理演算子

論理演算子は、論理式（評価結果が論理値となる式）を論理的に連結し、より複雑な論

理式を作る演算子である。

x && y // x かつ y が trueのとき
x || y // x または y が trueのとき
!(y) // x が false のとき

x and y // x && y と同じ
x or y // x || y と同じ
not y // !(x) と同じ

Konohaの論理演算子は、C/C++, Java互換の記法と、Python由来の記法の両方をサポー
トしている。どちらを使っても構わないが、C/C++, Javaとの互換性を重んじるなら、&&,
||, !を用いる方が無難である。

>>> a = 1
>>> b = 2
>>> a == 1 && b == 2
true
>>> a == b || a < b
true
>>> !(a == b)
true

論理演算を混在したとき、!, &&, ||の順に演算される。ただし、()で囲まれた式は優
先的に評価される。論理式を混在させるときは、読みやすさのため、()で明示的に優先

する論理式を与えるのがよい。

>>> a = 1
>>> b = 3

>>> !(a == 1) && b == 3 || a < b
true
>>> !(a == 1) || b == 3 && a < b
false
>>> (!(a == 1) || b == 3) && a < b
true

4.5.1 条件演算子

条件演算子は、条件によって評価する式を切り替える演算子である。Konohaは、次の
2種類の条件演算子をサポートしている。

x ? y : z // xが trueのとき y、そうでなければ z
x ?? y // xが nullのとき y

演算子 x ? y : z は、3 項演算子とも呼ばれ、条件式 x の評価結果によって、式 y

か式 zが評価される。3項演算子は、読みにくくなるので、各式の周りは ()で囲んだ方

がよい。

>>> a = 1; b = -1;
>>> a = (a < b) ? (a) : (b);
>>> a
-1

演算子 x ?? y は、C#で導入された null 検査用の条件演算子である。意味は、
(x != null) ? x : y と同じである。ただし、null 検査を連続するときは、こち
らの方がはるかに読みやすい。

>>> name = student["name"] ?? ninja["name"] ?? "unknown";

4.6 算術演算子

算術演算子は、2つの数値 x, yに対し、四則演算を行うときに用いられる。また、文字
列など一部のクラスには算術演算子は、連結や分離などの意味で定義されていることも

ある。

x + y // 加算
x - y // 減算
x * y // 乗算
x / y // 除算

x % y // 余り（モジュロ）
x mod y // x % y と同じ

モジュロ演算子 %は、C/C++, Javaで広く採用されているが、Konohaではフォーマッ
タ名と区別しにくいので、modを使うか、演算子の前後に明示的に空白を入れることが推

奨されている。

算術演算子は、四則演算の優先順位にしたがって評価される。つまり、乗算 (*)と除算
(/)の優先度は、乗算 (+)と除算 (-)より高く、同じ優先度の場合は左から順番に評価さ
れる。評価順序を変更する場合は、先に評価したい式を括弧 ()で囲めばよい。

>>> 1 + 2 * 3
7
>>> 1 + 2 - 3 * 4 / 2
-3
>>> (1 + 2) * 3
9

4.6.1 算術代入

算術代入は、代入演算子と算術演算子を組み合わせたシンタックスシュガーである。

x += y // x = x + y
x -= y // x = x - y
x *= y // x = x * y
x /= y // x = x / y

算術代入は、代入可能な右辺値であれば、全て適用することができる。

>>> a = [0, 1]
>>> a[0] += 1 // a[0] = a[0] + 1 と同じ
>>> a
[1, 1]

ただし、a[n++] += 1 のような副作用が生じる場合は、期待通りの算術代入にはな

らない。

4.6.2 インクリメントとデクリメント

Konohaは、”Everything is an Object”の世界で、Intオブジェクトは Immutableなので、
本来の意味でインクリメントとデクリメントをサポートできない。しかし、あまりに利用

者からの要望が多いため、シンタックスシュガーで対応している。現在のところ、前置、

後置演算の区別はない。

x++ // x = x.opNext()
++x // x = x.opNext()
x-- // x = x.opPrev()
--x // x = x.opPrev()

Konohaでは、インクリメント/デクリメント演算子は、ステートメントとして利用する
ことを推奨している。式として利用した場合、その評価順序は、将来にわたって保証され

ない。

>>> for(i = 0; i < 3; i++) print i;
i=0
i=1
i=2

4.7 ビット演算子

Konohaは、Cスタイルのビット演算子を Int型に対してサポートしている。スクリプ
ティング言語において、これらの演算子を多用することは稀であり、これらは教育用を主

目的*1として用意されている。

˜x // ビット反転
x & y // 論理積
x | y // 論理和
x ˆ y // 排他論理和 (XOR)
x << y // 左シフト
x >> y // 右シフト

フォーマッタ %bitsは、ビット演算子の結果を表示するとき利用できる。

>>> %bits(˜0)
11111111 11111111 11111111 11111111
>>> %bits(1 << 2)
00000000 00000000 00000000 00000100

*1 バーチャルマシンにおいて専用命令としてチューニングされないと意味である。



4.8 シーケンス演算子 27

4.8 シーケンス演算子

Konohaは、シーケンス（順序を保って並んだ集合）に対して、その操作に関する一連
の演算子を提供し、それらを総称してシーケンス演算子と呼ぶ。代表的なシーケンスの例

は、Array, byte[], Stringなどがある。

|s| // シーケンス s のサイズ
s[n] // n番目の値
s[n] = x // n番目に値 xをセットする
s[] = x // 全部に値 xをセットする
x in? s // s に xが含まれているか？
s << x // s への x の追加

シーケンス演算子では、シーケンスの要素の種類によらず、シーケンスのインデックス

n に対して、常に 0 ≤ n < |s| が保証される。(ただし、DictMap や HashMap のよう
に、インデックスの代わりにキーを用いる場合はこの限りでない。)

>>> a = [0, 1, 2, 3]
>>> |a|
4
>>> a[1]
1

4.8.1 インデックスの範囲

インデックス n が最大値|s|を超えた場合は、OutOfIndex!!例外となる。古い

Konohaでは、Pythonと同様に、最後尾から数えてインデックスし直していたが、配列の
境界チェックの最適化を考慮に入れて、負のインデックスは見送られた。

>>> a[-1]
** OutOfIndex!!:

4.8.2 シーケンスへの追加

演算子 <<は、特別なシーケンス演算子であり、複数の要素を連続して追加することも

できる。

>>> a = [1]
>>> a << 2 << 3 << 5 << 7;
>>> a
[1, 2, 3, 5, 7]

4.9 スライシング演算子

スライシング演算子は、シーケンスから部分シーケンスを取り出す演算子である。

Pythonのスライシング演算子とよく似ているが、Konohaでは部分シーケンスの定義の仕
方が 3種類用意されている。

s[x..y] // x番目から y番目 (含む)まで
s[x..<y] // x番目から y番目まで

// s[x..(y-1)] と同じ
s[x..+y] // x番目から y個の部分

// s[x..(x+y)] と同じ

文字列は、シーケンスの一種であり、スライシングで部分文字列を得ることができる。

>>> s = "konoha"
>>> s[2..3]
"no"
>>> s[2..<3]
"n"
>>> s[2..+3]
"noh"

よく似た演算子が 3 種類存在する理由は、これらはよく混同されて間違えるためであ
る。逆に、３種類、範囲除外 (..<)やオフセット (..+)をイメージしやすい演算子を導入
することで、どれかを正しく使いやすくなる。また、スライシング s[x..y]の範囲 x y

は、それぞれ省略可能である。省略したときは、それぞれ x = 0, y = |s| - 1 が初

期値として採用される。

>>> s = "konoha"
>>> s[2..]
"noha"
>>> s[..4]
"kono"
>>> s[..] // シーケンスのコピー
"konoha"

4.10 キャスト演算子

Konohaは、オブジェクト間のデータ変換や意味変換を強力にサポートするマッピング
機能を独自に導入している。これらは、C/C++, Java言語でおなじみのキャスト演算子に
統合され、利用可能である。

(T)x // xの型 Tへの変換（キャスト）
x to? T // xは型 Tへ変換可能か？

マップキャスト演算子 (T)xは、ソース型 typeof(x)とターゲット型 Tの関係で動

作が異なる。サブタイプの関係が成り立つ場合は、通常どおり、アップキャスト、もしく

はダウンキャストされる。それ以外の場合は、マッパーが検索され、それが評価される。

>>> Object o = "123";
>>> s = (String)o; // ダウンキャスト
>>> n = (int)o; // マップキャスト
>>> n
123

また、マッピングが定義されていない場合、つまりキャストができない場合は、

ClassCast!!例外となる。

>>> (int)true
** ClassCast!!: Boolean ==> Int

これらの演算子は、マッピング機能を応用して実現されている。詳しくは、「第??章
マッピング機能とデータ変換」で述べる。

4.11 イテレータ演算子*

イテレータ演算子は、オブジェクトから (標準)イテレータを得るときに用いる。foreach
文において、イテレータを生成するときに用いる（内部）演算子であるが、プログラマが

用いても構わない。

x.. // x の (標準)イテレータ

次の例では、整数配列 Int[]からイテレータ (Int..)を得ている。

>>> a = [0, 1, 2]
>>> typeof(a..)
>>> a..
1
2
3

イテレータ型や標準イテレータの定義は、「第??章イテレーション」で詳しく述べる。

4.12 セマンティック演算子**

Konohaは、オブジェクト指向モデルの軽量オントロジ拡張を行い、クラスやオブジェ
クト間の意味的な関係を比較することができる。

x isa? y // x は、y か？ IS-A 関係
x === y // x と y は意味的に等しいか？

Konoha は、文字列や数値に意味タグを付加することができる。次は、華氏 (32.0F) と
摂氏 (0.0C)の比較である。

>>> Float:F temp = 32.0F;
>>> temp == 0.0C;
false
>>> temp === 0.0C;
true

これらの演算子は、マッピング機能を応用して実現されている。詳しくは、「第??章セ
マンティックプログラミング」で述べる。

ステートメント

プログラムは、ステートメントの集まりである。本章では、Konohaの基本的なステー
トメントとその文法を説明する。あらかじめ断っておきたいが、Konoha は C/C++, Java
とステートメントレベルでの互換性を目標として設計されている。そのため、本章で紹介

するステートメントは、経験豊富なプログラマには読み飛ばしてもほとんど問題のない章

といえる。（頑張って書いた側に悲しい事実であるが、）なお、ステートメントの一般的な

訳語は、「文」である。本来、ステートメントに統一すべきかも知れないが、ifステートメ
ントと呼ぶのは冗長な気がするので、広く慣習的に用いられている「if文」的な呼称を用
いている。

5.1 式によるステートメント

もっとも簡単なステートメントは、プログラムの状態を変更する式 (expression) であ
る。代入式は、もっともよく使われるステートメントである。

s = "Hello " + name;
i *= 3;

インクリメント演算式 (++)、デクリメント演算式 (–)は、代入式の一種であるが、Konoha
ではステートメントとして利用することができる。

counter++;

関数/メソッドのコールも、もうひとつの代表的な式によるステートメントである。

System.out.println("hello,world");
Math.sin(0.5);

上記以外の任意の式もステートメントとして用いることができる。しかし、プログラム

の状態を変更しない式は、コンパイラによって無視され、実際には評価されない。

if(a == 1) print a;
a + 1; // 無視される
return a;

5.2 ブロック

(ステートメント)ブロックは、0個以上のステートメントまとめてひとつのステートメ
ントとして扱えるようにする。これは、任意のプログラムを中括弧 { }で囲むことで作

ることができる。ブロックの終端には、C時代からの慣習的にセミコロン (;)は必要ない。
また、ブロック内では、読みやすさのためインデントするのがよい。

{
x = Math.PI;
a = Math.cos(x);
print a;

}

注意：Konohaでは、メソッド (関数)内同一スコープであるため、ブロックの中で宣言
した変数はブロック外でも有効になる。詳しくは、「第 ??章関数」で述べられる。

5.2.1 空ステートメント

空ステートメントは、何もしないステートメントである。ブロックを用いて空ステート

メントを表すことができる。

{}

5.3 変数の型宣言

Konoha の変数は、静的に型付けされている。プログラマは、新しい変数を使うとき、
型宣言が必要となる。C/C++や Javaと同様、次の構文で変数 nameの型 T を宣言するこ

とができる。

T name;

また、変数宣言では、必要に応じて、代入演算子によって変数の初期値を与えることも

できる。また同じ型であれば、複数の変数をひとつのステートメントで宣言できる。

String s; // 初期値なし
String t = "hello,world"; // 初期値あり
String u, v = "ABC"; // 同じ型の複数の変数宣言

5.3.1 クラスと型

Konohaは、オブジェクト指向プログラミング言語であるため、全ての宣言されたクラ
ス Cを「C」型として利用することができる。また、クラス名 Cに修飾子を付けること
で、その型の null値の扱いを変更したり、配列型やイテレータ型の短縮名も利用すること
ができる。

C C 型
C? @Nullable C
C! @NonNull C
C[] C の配列型
C.. C のイテレータ型

Konoha の大きな特徴は、型に@Nullable または@NonNull を修飾できることであ
る。これらは、名前が示すとおり、変数の値として null値を認めるかどうかであるか、を
制約を付加したものである。Java との大きな違いは、何も修飾しなければ、原則として
NonNull型となる点である。この場合、変数宣言の初期値は、nullではなく、そのクラス
のデフォルト値となる。

>>> String s; // 初期値のデフォルト値
>>> s
""
>>> s = null // Null例外が発生
** Null!!

注意：各クラス Cのデフォルト値は、組み込み関数 default(C)を用いて得ることが

できる。また、デフォルト値の中には Nullオブジェクトパターンにしたがって、null値
のように振る舞う値もある。

逆に、C#風に?を型名に付けるか、@Nullable修飾子を用いれば、nullを保持可能な変
数となる。この場合の初期値は、nullである。

>>> String? t; // 初期値は null
>>> t
null

5.3.2 var 文*

var文は、JavaScriptなど一部のスクリプティング言語で採用されている変数宣言のた
めの特別なステートメントである。Konoha では、静的に型付けして型宣言を行うため、
var文では初期値から型推論を行い、静的な型付けを行っている。

var i; // Any 型（型推論なし)
var s = "hello,world"; // s は、String 型
var x = Math.cos(0.75); // x は、Float 型

注意：var文は省略可能である。始めての使用する変数への代入は、型推論による変数
宣言とみなされる。詳細は、「第 refdynamic章ダイナミック・スタイル」で解説する。

5.4 if/else 文

if文は、もっとも基本的な制御ステートメントである。与えられた条件式に応じて、続
いて実行するステートメントの切り替えるときに使用する。if文には、2種類の形式があ
る。次は、簡単な形式である。

if (expr) stmt

この形式では、条件式 expr がまず評価される。その結果が、trueであれば、それに続
くステートメント stmtが実行される。もし条件式の結果が falseであれば、ステートメ
ント stmtは実行されない。

>>> n = 4
>>> if (n mod 2 == 0) print "even";
even

if文に続くステートメントとして、ブロックを用いることもできる。ブロックを用いれ
ば、0個以上のステートメントを順次実行することもできる。

>>> if (n mod 2 == 0) {
... print "even";
... }
even

5.4.1 else節

if文に続く else節は、別のステートメントを与えるときに利用する。

if (expr) stmt1 else stmt2

この形式の if文では、まず条件式 exprが評価され、もしそれが trueであればステート
メント stmt1が実行され、そうでなければ stmt2が実行される。

>>> n = 3
>>> if (n mod 2 == 0) {
... print "even";
... } else {
... print "odd";
... }
odd

5.4.2 else if

if/else文を使うことで、2つのステートメントを条件に応じてディスパッチすることが
できた。更に、if/else文を組み合わせ、あたかも else if文のように用いると、3つ以上の
条件分岐をすっきりと書くことができる。

if(n == 1) {
// n が 1のとき

}
else if(n == 2) {
// n が 2のとき

}
else if(n == 3) {
// n が 3のとき

}
else {

// それ以外のとき
}

else if文は、特別なステートメントではなく、ふつうの if/else文の組み合わせたに過
ぎない。条件式は、(n == 1)から順番に評価されていくため、後に続く条件判定の処

理ほど遅くなる。

5.5 switch文**

switch 文は、多重ディスパッチを行うための専用のステートメントである。文法は、
C/C++や Javaと同様に次のとおりである。式 expr の評価結果にマッチする sf case節が
選択され、そのステートメントが実行される。

switch (expr) {
case c1 : stmt1;
case c2 : stmt2;
...

case cn : stmtn;
default : stmt;

}

sf switch文を用いると、一般に　 if/else文よりもすっきりと複雑な条件分岐をかくこ
とができる。次は、sf else if の節で用いた多重分岐の sf switch バージョンである。注意
すべき点は、sf break文である。これは、各 sf case節は、条件分岐の始まりを表すに過ぎ
ず、明示的に分岐の終端を表している。

switch(n) {
case 1 :
// n が 1のとき

break;
case 2 :
// n が 2のとき

break;
case 3 :
// n が 3のとき

break;
default:
// それ以外のとき

}

Konohaと Cや Javaとの違いは、sf case節のマッチングにある。Cや Javaは、sf case
節では整数リテラルのみマッチングする条件として与えることができたが、Konohaでは
任意のリテラルを与えることができる。（ただし、式は与えることができない。）

switch(person.type) {
case "student" :
case "teachter" :
// 学割?

break;
case "officer" :
// 公務員割?

break;
default:
// それ以外のとき

}

5.6 while/do-while 文

while文は、あるステートメントを繰り返し実行させる基本的なステートメントである。
ループの制御構造とも呼ばれる。その文法は次のとおりである。

while (expr) stmt

while 文は、条件式 expr が true の間、続くステートメント stmt を繰り返す。次は、

変数 aが 0より大きい間、繰り返し実行される。

a = 1;
while(a > 0) {

a = Int.random(10);
print a;

}

Konohaでは、ループを実現するステートメントとして、do-while文のバリエーション
がある。

do stmt while (expr);
do-whileループは、先にステートメント stmtを一度実行したあと、条件 expr によっ

て繰り返しの有無を定を行う。whileと do-whileの違いは、whileが条件に次第では 1回
もステートメントが実行されないのに対し、do-while の方は必ず１回は実行される点で
ある。

do {
a = Int.random(10);
print a;

}while(a > 0);

5.7 for文

for文は、変数の初期化とその値の増減を繰り返すときに便利な while文の置き換えで
ある。実際、次のソースコードは、

for(i = 0; i < 10; i++) {
print i;

}

以下の while文と全く同じである。

i = 0;
while(i < 10) {

print i;
i++;

}

5.8 break 文と continue 文

Konoha は、break と continue という 2 種類のループ制御のステートメントをもって
いる。

break 文を使うと、現在繰り返しているループ構造のブロックから抜け出すことがで
きる。

while(true) {
dice = Int.random(6) + 1;
dice2 = Int.random(6) + 1;
print dice, dice2;
if(dice == dice2) break; // ゾロ目の場合、終了

}

continue 文を使うと、現在繰り返しているループ構造のブロックの残りのパートをス
キップする。

for(i = 0; i < 10; i++) {
if(i % 2 == 0) continue;
print i, ...;

}

多重ループからの breakと continue
ループ構造は、while 文や for/foreach 文を組み合わせて、多重ループを作ることがで
きる。多重ループの中では、どのループを対象として、breakしたり、continueするのか
問題となります。

Konohaでは、ループにラベルをつけることで、どのループからの break/continueなの
か明示的に示すことができる。また、ラベルを省略すれば、今まで通り暗示的に最も内側

のループが対象となる。

OUTER:
for(y = 0; y < 8; y++) {

INNER:
for(x = 0; x < 8; x++) {

if (x == y) continue OUTER; /* 外側 */
if (x < y) break; /* 内側 */
print x, y;

}
}

5.9 for/in文

foreachは、イテレータ内の各オブジェクトに対して繰り返し処理を行う専用のループ
構造である。多くの場合、for文よりもわかりやすく、高速な処理が可能になります。

foreach(String line from new InputStream("file.txt")) {
print line;

}

All of objects in Konoha are convertible to an Iterator object. See the upcoming ”Foreach
and Iteration ” section for detailed information.

Konohaは、foreach文を大きく拡張してあります。詳しくは、「第??節 foreachとクエ
リー処理」を参照してください。

数値

コンピューティングプロセッサ (CPU)は、コンピュータシステムの中心部であり、様々
な形式の整数や浮動点小数の演算をサポートしている。プログラミング言語は、これらを,
例えば C言語の場合、char, short, int, long long long int, float, double, long doubleの
ように、異なる型で区別して利用できるようになっている。一方、Konohaは、プロセッ
サの性能やメモリ効率を最大利用するより、プログラミングのしやすさに力点を置いてい

る。そのため、最も使いやすい整数と浮動小数点の形式を選んで、Intと Floatクラスとし
て箱詰め (boxing)している。本章では、これらのクラスを用いた数値の扱いを説明する。

6.1 数値と型

Konohaは、２種類の数値表現をサポートしている。これらは、C言語における 64ビッ
ト符号付き整数 (long long int)と倍精度浮動小数点数 (double)に相当し、日常的な利用に
おいては自然数や実数を表現するのに十分な値域と精度を持っている。

Konohaでは、“Everything is an Object”の設計であり、数値を含め、すべての値はオブ
ジェクトとなっている。Intクラスと Floatクラスは、それぞれ整数値と浮動小数点数値
を箱詰め (boxing)してインスタンス化するためのクラスである。

>>> (1).class
Int

注意：「箱詰め (boxing)」の話だけをすると、プログラミング言語実装に詳しい方は不
安になるだろう。Konohaの内部をより正確にいえば、オブジェクトとしてアクセスする
ときのみ、boxingされており、それ以外は、もちろん unboxingされた状態で数値処理さ
れている。不必要なメモリ消費は最低限に抑えられている。

6.1.1 int, float型

Konoha では、クラス名は英大文字で始まり、そのまま型名として用いられる。この
ルールにしたがえば、整数値、浮動少数点値を用いるための型は、Int, Float型となる。し
かし、C/C++, Javaなど、ほとんどすべてのプログラミング言語において、これらの型は
int, floatのように英小文字で始まっている。
ここで、言語設計の統一的美しさにこだわれば、Konohaのルールにしたがってもらえ
ばよいものであるが、「言語」というものは実際に使っている人たちが主人公である。プ

ログラミング言語も、そろそろ半世紀の歴史をもつわけであるから、言語設計者が好き勝

手に文法を決められる時代はもう終わったと考えてよいだろう。統一的な美しさはない

が、Konohaでは、intや floatも別名定義し、プログラム中で利用することができる。
最後に、Konoha 風の Int と伝統的な int のどちらを使った方がよいかという問題が残
る。Konoha のローカルルールに固執して Int/Float を使うより、C/C++ や Java などと
ソースコードの交換性を保つため、int/floatを使う方がよいと考える。また、同じ理由で
longや double型も定義されており、Int/Floatの別名として定義されている。

6.1.2 Decimal**

Konohaプロジェクトは、将来、小数点以下の精度を保持した無限長の数値をサポート
することを計画している。

6.2 数値リテラル

数値は、ふつうに表記すれば、予約語や識別子と区別がつかないことはない。数値リテ

ラルと言っても、極めて自然な表記であるが、整数と浮動小数点を区別するため、表記上

のルールがある。

6.2.1 整数リテラル

スクリプティング言語では、short や long などサイズの異なる整数を使い分ける可能
性は少なく、単純に自然数を表すために整数を用いることが多い。Intクラスでは、数値
の上限下限を最も気にしなくて済む 64ビット整数*1を採用している。

Konoha では、10 進数の整数は、数字の列として書ける。整数の値の範囲は、-
9,223,372,036,854,775,807 から 9,223,372,036,854,775,807 となる。つまり、日常の利
用ではまず足りなくなる心配のない値域をカバーしている。

*1 KNH_USING_INT32オプションを用いれば、32ビット整数に変更することも可能である。

0
3
100000000

10進数表記に加えて、Konohaは、16進表記と 2進表記をサポートしている。16進表
記の整数リテラルは、0xもしくは 0Xで始まり、続いて数字 (0-9)もしくは 10から 15ま
での数字を表す英字 (A-F)が続く形式である。

0xff // 15 * 16 + 15 = 255
0xCAFE911

クラスルームでは、結構、2進数の説明も頻出する。そのため、Konohaは独自の 2進
数リテラルとして、0bもしくは、0Bから始まり、0もしくは 1が続く 2進数で整数を与
えることができる。

0b1111 // 0xff = 255

6.2.2 Float リテラル

浮動小数点数 (floating-point numbers)は、全て Floatクラスの値として表現される。そ
の名前から想像されるものとは異なり、IEEE 745 標準で定義された 64 ビット倍精度形
式を用いている。これらは、C/C++, Java言語では、通常 double型で表現される形式で
ある。

Floatリテラルは、小数点を必ず含む。通常、整数パートの数列に始まり、小数点、続
いて小数点以下の数列が続く形式である。また、eもしくは Eによって 10n 乗形式の位を

追加することも可能である。

3.14
2345.789
.33333333333
6.02e23
1.473E-32

6.2.3 桁取り表記

我々は、日常、大きな数字を間違えることなく扱うため、1000桁ごとにカンマ (,)を入
れて表記することが多い。プログラミング言語では、カンマは特殊な意味をもった演算子

の一種として扱われるが、Konohaでは、アンダースコア (_)を数値リテラルの任意の箇
所に入れることが認められ、カンマの代わりに桁取り表記も可能となる。OCaml風の拡
張であるが、ちょっとだけ読みやすくなる。

1_0000
9_223_372_036_854_775_807

6.2.4 意味タグ拡張と単位*

Konohaは、セマンティックプログラミングの機能として、Intや Floatクラスに対して
意味タグを付加することができる。意味タグは、単位など直感的なタグを用いることが可

能であり、数値リテラルの後置辞として利用することができる。詳しくは、「第 ??章セマ
ンティックプログラミング」で解説する。

32.195km // Float:km
80[km/h] // Int[80/km]

注意：意味タグが定義されていないときは、意味タグは無視されて、通常の整数、浮動

小数点数のリテラルとして扱われる。型チェックはおこなわれないが、読みやすさのため

に意味タグを使っても構わない。

6.3 数値演算

Konohaでは、数値はクラスによって箱詰めされているが、その値はそのままネイティ
ブの数値表現、C言語で言うところの long long intと doubleである。したがって、その
数値演算の振る舞いは、C言語の演算子と同じである。特に、浮動小数点演算において計
算結果が微妙に異なることがバグレポートとして報告されるが、そもそも浮動小数点演算

というのは桁落ち、丸め誤差などが含まれていることを思い出して欲しい。

6.3.1 四則演算

Int, Floatは、それぞれの型の範囲において四則演算 (+,-,*./)を定義している。つまり、
Int型間の演算では Int型の結果がえられ、Float型間の演算では、結果は Float型になる。

>>> 7 / 2 // Int型の除算
3
>>> 7.0 / 2.0 // Float型の除算
3.5000000

一方、Int型と Float型を混在して演算したときは、Int型の値は全て自動的に Float型
に変換されたのち、演算される。したがって、結果は Float型となる。

>>> 7 / 2.0 // 7は 7.0へ
3.5000000
>>> 7.0 / 2 // 2は、2.0へ

3.5000000

ゼロ除算の場合は、Int, Floatともに、Arithmetic!!例外が通知される。

>>> 1 / 0 // ゼロ除算
Arithmetic!!: Zero Divide

優先順位

Konohaの四則演算は、日常の数学の演算と同じ順序で計算される。つまり、乗算と除
算は、加算と減算に優先して実行される。演算の優先順位を変更するときは、優先対した

い演算を ()で囲む。

>>> 1 + 2 * 3
7
>>> (1 + 2) * 3
9

Konohaでは、四則演算以外にも数多くの演算子をサポートしている。これらを混在し
て利用するときは、優先する演算子を ()で囲む習慣をつけると、思わぬ誤動作を防ぎ、ま
た読みやすさも向上する。

6.3.2 ビット演算

Konohaは、Int型の数値のみ、ビット演算をサポートしている。

>>> %bits(1)
00000000 00000000 00000000 00000001
>>> %bits(˜1)
11111111 11111111 11111111 11111110
>>> %bits(2)
00000000 00000000 00000000 00000010
>>> %bits(1&2)
00000000 00000000 00000000 00000000
>>> %bits(1|2)
00000000 00000000 00000000 00000011
>>> %bits(1<<4)
00000000 00000000 00000000 00010000
>>>

スクリプティング言語では、低レベルな処理を書く機会が少ないため、ビット演算子

を活用する機会は少ない。さらに、Konohaでは、ビット演算子の記法は他のよく使われ
るオペレータ (|a|y や OUT << "hi"など) に転用されているため、演算子の優先度が
C/C++, Javaと異なることがある。この違いを覚えるよりは、ビット演算子と四則演算子

を組み合わせるときは、()で囲んで優先する演算子を明示した方がよい。

>>> %bits(1+1<<2) // 優先度が？
00000000 00000000 00000000 00001000
>>> %bits(1+(1<<2))
00000000 00000000 00000000 00000101
>>> %bits((1+1)<<2)
00000000 00000000 00000000 00001000

Float の数値を IEEE 754 浮動小数点のビットレイアウトにしたがってビット演算を
したいときは、まず Float.floatToIntBits() を用いて Int 型へ変換したのち、
Float.intBitsToFloat() で Float 型に変換することで可能になる。ちなみに、
Konohaは 32ビットシステムであっても、Intと Floatはともに 64ビットで表現される。

>>> n = (float)1.0; // 数値変換
>>> %bits(n)
00000000 00000000 00000000 00000001
>>> f = Float.floatToIntBits(1.0) // ビットレイアウト維持
>>> %bits(f)
00111111 11110000 00000000 00000000

6.3.3 比較演算

Int 型, Float 型は、ともに比較演算子 (==, !=, <, <=, >=, >) によって数値を比較でき
る。また、Int型、Float型の比較を混在して用いたとき、Konohaは通常、異なる型の比
較はオブジェクト IDによる比較になるが、特別に自動的に Intクラスを Floatクラスに変
換して比較を行う。

>>> 1 < 2.0
true
>>> 3.2 != 3
false

Int型の比較演算子は、実用上問題がおこらないが、Float型の場合は、IEEE754規格に
基づいた表現のため、実用上は同値であっても同値とみなされない。そこで、(正規表現
のための)マッチング演算子=˜を用いると、簡単にほぼ等しいを演算することができる。
有効桁数は、小数点以下 5桁である。

>>> 1.00003 == 1.0
false
>>> 1.00003 =˜ 1.0
true

6.3.4 数値変換

Konohaでは、Int型と Float型の型変換は、必要に応じて自動的に行われる。これらを
明示的におこないたい場合は、キャスト演算を用いることができる。ただし、Konohaの
数値変換の振る舞いは、C言語の数値変換の振る舞いに基づいている。特に、Floatから
Intへの変換は、切り捨てになるため注意が必要である。

>>> (int)1.8
1
>>> (int)-1.8
-1

,小数点以下 n桁で切り上げ、四捨五入、切り下げを行いときは、mathパッケージのク
ラス関数 Math.ceil(), Math.round(), Math.floor()を用いる。

>>> using math.*;
>>> Math.ceil(1.8)
2.0000000
>>> Math.round(1.8)
2.0000000
>>> Math.floor(1.8)
1.0000000

6.3.5 文字列との変換

Konohaは、全てのクラス間の変換がキャスト演算子に統合されている。文字列への変
換も文字列からの変換もキャストで行うことができる。ただし、数値の書式を指定したい

場合は、次節のフォーマッタを用いる。

>>> (String)1.8
"1.8000000"
>>> (int)"123"
123

文字列から数値への変換は、数値リテラルの文法も解釈される。

>>> (int)"1"
1
>>> (int)"0xff"
255
>>> (float)"6.0221415e23"
602214149999999896780800.000000

数値以外の文字列は、それぞれ Int型、Float型のデフォルト値へ変換される。デフォル
ト値の代わりに（変換に失敗したことを知るために）nullを得たいときは、Nullableキャ
ストを用いることができる。

>>> (int)"hello,world" // 数値でない
0
>>> (int?)"hello,world" // 数値でない
null

6.4 数値フォーマッタ

数値フォーマッタは、C言語 printf書式に由来しているものが多いが、独自に定義

されたものは %bitsなどいくつか存在する。ここでは、代表的な数値フォーマッタにつ

いて紹介する。

6.4.1 10進書式 %d

10進数 (digit)書式では、文字列の幅 mを %mdのように指定できる。また、%0mdよ

うに書式を与えると、空白の代わりに 0で埋められる。（これは、数値としての順序が文
字列化したときも保持されるため、よく用いられる。）

>>> %d(123)
"123"
>>> %4d(123) // 幅 (4)を指定
" 123"
>>> %-4d(123) // 左寄せ
"123 "
>>> %04d(123) // 0パディング
"0123"

6.4.2 16進書式 %x

10進数 (digit)書式では、10進書式同様に、文字列の幅mを %mxのように指定できる。

>>> %x(123)
"7b"
>>> %8x(123) // 文字列の幅 8
" 7b"
>>> %08x(123) // 0パディング
"0000007b"

6.4.3 小数書式 %f

小数書式は、Floatクラスだけでなく、Intクラスもサポートしている。したがって、整
数も型変換することないしに、小数書式でフォーマッティング可能である。

>>> %f(1)
1.000000
>>> %f(1.0)
1.000000

小数書式では、全体の幅mと小数点以下の桁数 nを %m.nfのように指定できる。

>>> %f(3.14159)
"3.141590"
>>> %6.2f(3.14159) // 小数点以下 2桁
" 3.14"
>>> %6.3f(3.14159) // 小数点以下 3桁
" 3.142"

6.4.4 ビット書式 %bits

ビット書式は、printf書式に存在しない独自の書式である。整数や浮動小数点数のビッ
トレイアウトを確認するために導入されており、クラスルーム用途の書式である。

>>> %bits(-1) // 2の補数表現 :)
11111111 11111111 11111111 11111111

6.4.5 UCS4文字書式 %cn

文字書式は、printf書式でもおなじみの書式であり、整数から文字コードを用いて文字
に変換してくれる。ただし、Konoha言語では、Unicode文字コード (UCS4)を用いてい
るため、ASCII文字コード以外の文字も変換することができる。

>>> %c(30) // ASCII (<128)
"A"
>>> %c(0x3042) // UCS4 (>128)
"あ"

6.5 乱数生成

乱数生成は、Intと Floatクラスの値をランダムに生成する一種のコンストラクタであ
る。Konohaでは、標準乱数生成器として、統計処理分野で愛用される高品質な疑似乱数
アルゴリズムMersenne Twister [1]を採用している。
乱数生成は、クラス関数として提供されている。Int.random(n) は、0 から n(含
まれない)までの整数乱数を生成し、Float Float.random()は、浮動小数点乱数 x
(0.0 < x < 1.0)を生成するクラス関数である。
例えば、さいころ (1 6)を作りたいときは、次のように用いることができる。

>>> Int.random(6) + 1 // さいころ
3
>>> Int.random(6) + 1 // 違う目が出た
6

6.5.1 乱数生成器の初期化

Konohaは、起動時の時刻とラインタイム情報から起動する度に乱数生成器を初期化し
ている。そのため、通常、乱数生成器を初期化する必要はない。ただし、明示的に乱数生

成器を初期化したいときは、System.setRandomSeed(seed)を用いることで、再初

期化可能である。(パラメータ seedに対し、0を与えたときのみ、その時刻とランタイム
情報からシードが生成される。)

>>> System.setRandomSeed(1)　 // 初期化
>>> Int.random() // 初期化後は同じ　
1234794094773155764

注意：同じ seed値で初期化すると、乱数生成系列が同じになる。つまり、毎回、同じ順
番で同じ乱数生成が繰り返される。ただし、Konohaの乱数生成器は、複数の言語エンジ
ン/スレッドから共有されているため、マルチ言語インスタンスやスレッド実行下での乱
数生成は、逆に同一性は保証されなくなる。



51

第 7章

文字列

文字は、人類史の始まり以来、最も重要な情報表現の手段である。情報社会といわれる

現在においても、社会や文化を支える主要な情報は、文字によって記録され、伝えられて

いる。そのため、プログラミングにおける文字の重要性も明らかといえる。文字やテキス

トをうまく扱うことは、情報処理の中心的な役割を占めているからである。

プログラミング言語では、文字を１文字、1文字扱うのは不便なので、複数個以上の文
字の並びを「文字列 (string)」として扱う。Konohaは、文字列を表現するため、Stringク
ラスを提供し、独自のフォーマティング機能とあわせて、自由自在な文字列操作を実現し

ている。

7.1 文字列と型

ほぼ全てのプログラミング言語は、海外で生まれたという経緯もあって、長らく「文字

=ASCIIコード=1バイト」という前提で開発されてきた。これが不幸の始まりで、スクリ
プティング言語であっても、日本語を処理する場合はどうしても１文字が何バイトに相当

するか意識しながら文字列処理を強要されることが多かった。

Konohaは、21世紀になってからの設計であり、Unicode文字を採用し、1文字はそれ
が何バイトであっても 1文字と数えている。つまり、文字列処理において、文字コードを
意識しなくて済む設計になっている。

文字列は、0個以上の Unicode文字からなる。Konohaでは、文字単体を扱う特別な型
は存在せず、文字は長さ１の文字列として表現される。

"a" // ASCII文字
"あ" // Unicode 文字

どうしても文字コードを扱う必要があるときは、UCS4(31 ビット) コードとして、Int
型を用いて扱うことができる。

>>> ("a").getChar() // ASCIIコード

>>> ("あ").getChar() // UCS4 コード

Konohaは、Stringの内部エンコーディングとして、UTF8を利用している。本来、文
字列は実装依存しない抽象化されたデータ構造であるべきであるが、UTF8以外はバイト
列 (Bytes/byte[])して扱うことになる。Stringから、各種エンコーディングにしたがって
バイト列に変換する方法も用意されている。

s.getBytes("shift_jis") // Shift_JIS へ変換
new String(byte, "shift_jis");

注意：入出力ストリームにはエンコーディングが設定できるようになっているため、エ

ンコーディングを処理するケースはあまりない。n

7.2 文字列リテラル

プログラミング言語では、「プログラム要素の識別子」と「データとしての文字列」を

区別するため、文字リテラルのルールにしたがって、文字列をデータ値として記述する。

Konohaは、C/C++、Python、C#に由来する 3種類の文字列リテラル、さらに独自のリテ
ラルもサポートしている。ここでは、これらを順番に説明する。

7.2.1 ダブルクオート

最も基本的な文字列リテラルは、文字列をダブルクオート (”)で囲む形式である。この
リテラルは、C言語を含め、ほぼ全てのプログラミング言語で広く採用されている。

"" // 空文字列
"hello, world"
"3.14"
"Say \"hi.\"" // エスケープ記号
"いろは" // Unicode 文字

文字列リテラルとして、そのまま表現できない文字、例えば改行や"は、エスケープ記

号 (\)を用いて記述する。Konohaは、次の 6種類のエスケープシーケンスのみサポート
する。

sequence Character represented
\n 改行 (LF)
\t タブ
\\ \記号
\’ ’シングルクオート (""中)

\" "ダブルクオート (’’中)
\: :コロン (’’中)

7.2.2 シングルクオート

Konohaでは、シングルクオート (’)も、コロン (:)の扱いに注意すれば、文字列リテラ
ルの始まりと終わりとして利用することができる。シングルクオートを用いると、ダブル

クオート (”)を含む文字をすっきりと表すことができる。

’a’ // "a"と同じ
’Say "hi"’ // "Say \"hi.\"" と同じ
’http\://’ // : はエスケープが必要

シングルクオートは、正式には「タグ付き文字列」のリテラルとして定義されている。

タグ付き文字列とは、文字列の先頭に’tag:...’のようなタグを付けて、文字列の意味

を識別し、その意味にしたがって適切なオブジェクトを生成するリテラルである。

’re:s$’ // 正規表現 (Regex)
’file:/tmp/konoha.k’
’isbn:978-4-87311-329-6’
’IO!!:File Not Found’ // 例外 (Exception)

注意：「タグ付き文字列」のタグが未定義な場合は、通常の文字列リテラルとして解釈

される。ただし、明示的にタグでないことを示すためには、エスケープ (\:)する必要が
ある。詳しくは、「第??章セマンティックプログラミング」で述べる。

7.2.3 バッククオート

バッククオート (‘)も、%記号の扱いに注意すれば、文字列リテラルの始まりと終わり
の記号として利用できる。こちらを用いれば、シングルクオートとダブルクオートが混在

する文字列をすっきりと表すことができる。

‘Naruto said, "I’m a ninja."‘

バッククオートは、正式には「インライン・フォーマッティング」、つまり文字リテラ

ル中の書式が評価される特別な文字列リテラルとして用いられる。次は、%d{}内の式が

評価され、%dフォーマッタで整形された文字列が埋め込まれている。

>>> core =2
>>> ‘make -j%d{core+1}‘
"make -j3"



54 第 7章 文字列

もしバッククオートを文字リテラルとして利用するときは、% 文字は、\%、もしくは

%% のようにエスケープ処理が必要になる。詳しくは、「第 7.6 節 インラインフォーマッ
ティング」で述べる。

7.2.4 トリプルクオート

トリプルクオートは、Pythonに由来するテキスト用の文字リテラルである。”’もしく
は、”””で囲まれたテキストは、改行やタブ文字もそのまま文字列として扱われる。ただ
し、エスケープシーケンスも有効である。

’’’
This is a rather long string containing
several lines of text just as you would do in C.

Note that whitespace at the beginning of the
line is significant.

’’’

また、トリプルバッククオート (“‘) によるインラインフォーマッティングも有効で
ある。

‘‘‘
configure
make -j%d{core+1}
‘‘‘

Konohaでは、読みやすさを向上するため、ちょっとした改良をのトリプルクオートに
加えてある。トリプルクオートの先頭が改行の場合は、その改行は無視される。したがっ

て、上記のトリプルクオート文字列は、”c”から始まる。

7.2.5 RAW 文字列

RAW 文字列とは、エスケープシーケンスを無視した文字列リテラルである。Konoha
は、C#に由来する記法を採用し、文字リテラルの前に@を付加すると RAW文字列とし
て用いられる。

@"C:\My Document\Programming\Konoha"

7.2.6 行リテラル

行リテラルは、ドキュメントや他言語のスクリプトを読みやすく書くための、 Konoha
独自の拡張リテラルである。 #で始まるトークンは、その行の終わりまでをそのまま文字

列として解釈される。また、途中で、#が入っていてもそれは文字列の一部として解釈さ

れる。ただし、#直後の空白 (スペース)だけは、読みやすさのため挿入と解釈され、それ
は無視される。行リテラルは、バッククオートと同じく、「インライン・フォーマッティ

ング」を解釈する。

# make -j3 　　 // "make -j3"と同じ
# make -j%d{core+1}

行リテラルは、文字列リテラルの連結のところで述べるとおり、ドキュメントや他言語

のスクリプトを（読みやすく）書くための記法である。多くのスクリプティング言語で

は、#はコメントと解釈されるので、多用すると読みにくくなる。次のソースは、Konoha
の文法的に正しい用例であるが、読みにくくなる。

>>> s = # make -j3

7.2.7 文字列リテラルの連結

文字列リテラルは、複数並べると、自動的に連結されてひとつの文字列となる。異なる

タイプのクオートを混在させたときは、先頭のクオートタイプで連結される。

"hello" "," "world"　　 // "hello,world"と同じ

文字列リテラルの間に改行があると、連結するとき自動的に改行を含めて、連結され

る。これは、行リテラルの連結のときに、特に威力を発揮する。

Konohaでは、Groovyと同じく、メソッドや関数の第一パラメータに文字列リテラルを
書くとき、関数コール演算子 ()を省略することができる。行リテラルの連結と組み合わ

せて利用すると、すっきりとスクリプトが書くことが可能になる。

db.query
# create table book (
# name char[20];
# title char[80];
# );

下記のソースと完全に同じである。

db.query(’’’
create table book (
name char[20];
title char[80];
);’’’)

7.3 文字列と演算子

文字列は、「文字」配列である。ただし、Konoha の String は、不変 (Immutable) オブ
ジェクトであるため、Read Onlyの「文字」配列となる。

7.3.1 長さとシーケンス

文字列は、0個以上の文字が並んだシーケンスである。Konohaでは、シーケンス sは、
|s|演算子を用いることで、その長さを得ることができる。文字列の場合、そこに含まれ

る文字の数である。

>>> s = "";　　
>>> t = "ABC";
>>> u = "ABCいろは";
>>> |s|
0
>>> |t|
3
>>> |u|
6

Konohaの文字列の長さは、多国語化された設計となっているため、文字のバイト数に
関わらず、文字数が文字列の長さとなる。ASCII(1バイト)文字列を前提としたバイト数
でないことに注意したい。もし文字列のバイト数をしりたい場合は、一旦、バイト配列

(byte[])に変換すればよい。ただし、このとき UTF8エンコーディングが用いられ、そ
れにしたがい日本語は 3バイト以上になっている。

>>> s = "ABCいろは";
>>> b = (byte[])s;
>>> |b|
12

文字列は、文字の配列として扱うことができる。配列のインデックスは、0から始まり、
|s| - 1である。

>>> s = "ABCいろは";
>>> s[0]
"A"
>>> s[|s|-1] // 最後の一文字
"は"

注意：Konohaの文字列は、変更不能 (immutable)である。通常の配列のように値を変
更することができない。

文字列から部分文字列 (substring)を取り出すときは、シーケンススライス演算子を用い
ることができる。

>>> s = "ABCいろは";
>>> s[..2]
"ABC"
>>> s[3..]
"いろは"
>>> s[2..4]
"Cいろ"

7.3.2 連結

文字列の連結は、Javaと同様に、加算 (+)演算子によって書くことができる。この加算
は、もちろん結合則や交換則に基づくものではないが、操作が直感的であり、広く使われ

ている。

>>> "Uzumaki" + " " + "Naruto"
"Uzumaki Naruto"

Konoha は、文字列の加算を最も優先されるオペレータとして定義している。つまり、
文字列と文字列以外のオブジェクトを混在して換算した場合、全て文字列に変換されたの

ち連結される。

>>> "Naruto " + 8 + 1 // %s(8) + %s(1)と同じ
"Naruto 81"

もし数値演算を優先したい場合は、括弧 ()で囲んで先に計算する。

>>> "Naruto " + (8 + 1) // %s(8 + 1)と同じ
"Naruto 9"

7.3.3 連結、除去、分割

Konohaでは、加算に加えて、減算、乗算、除算の演算子にもそれぞれ各演算子の意味
に近い文字列操作として定義されている。これらも数学的な性質は継承しないが、文字列

処理としてときに便利である。

文字列の減算 s - tは、文字列 sから指定された文字列 tを取り除く操作である。こ
れは、s.replace(t, "")に相当する。

>>> "Uzumaki Naruto" - "a" - "u";
"Uzmki, Nrto"

文字列の乗算 s * n は、文字列の四則演算の中で唯一整数値を引数にとる。しかし、

そこから得られる結果は極めて直感的に、文字列 sを n個連結したものである。

>>> "Naruto" * 3;
"NarutoNarutoNaruto"
>>> "hello" * 0
""

文字列の除算 s / t は、文字列 s を文字列 t で分割したときの先頭の文字列である。
文字列の余算 s % tは、分割された残りの文字列となる。これは、s.split(t)の簡

易版として利用される。

>>> "Uzumaki Naruto" / " "
"Uzumaki"
>>> "Uzumaki Naruto" % " " // mod のこと
"Naruto"

7.3.4 文字列の比較

文字列は、関係演算子によって比較することができる。そのとき、単純にアルファベッ

トオーダー (UTF8のコード順)で比較され、英大文字/英小文字は無視されない。

>>> "Uzumaki" == "Naruto"
false
>>> "Uzumaki" > "Naruto"
ture

7.4 文字列メソッド

Konohaは、Java言語に由来する Stringメソッドを標準的にサポートし、更により多
くの文字列操作メソッドも拡張している。本節でとりあげるメソッドは、その一部であ

る。しかし、本節のメソッドだけで、Konohaプロジェクトで利用しているドメイン言語
(DSL)と Cソースコード生成器を作成することができる。

7.4.1 startsWith(), endsWith()

Konoha は、正規表現によるパターンマッチングが利用できるが、多くの場合、文
字列の先頭、もしくは末尾のみマッチングするだけで事足りることが多い。メソッド

s.startsWith(t) は、文字列 s の先頭が文字列 t と一致するか判定するときに利用

できる。逆に、メソッド s.endsWith(t)は、文字列 sの末尾が文字列 tと一致するか

判定できる。

>>> s = "Uzumaki Naruto"
>>> s.startsWith("Uzumaki")
true
>>> s.startsWith("Naruto")
false
>>> s.endsWith("Uzumaki")
false
>>> s.endsWith("Naruto")
true

startsWith()と endsWith()は、英大文字/小文字を区別する。equals()と同
様に IgnoreCaseバージョンを利用すれば、英大文字/英小文字を無視してマッチングを行
うことができる。

>>> s = "Uzumaki Naruto"
>>> s.startsWith("uzumaki")
false
>>> s.startsWith:IgnoreCase("uzumaki")
true

7.4.2 検索: indexOf(), lastIndexOf()

index()系の文字列操作は、C言語の標準ライブラリから存在する古典的な文字列ラ
イブラリである。Konohaでは、Java同様に、文字でなく、部分文字列の検索ができる。
メソッド s.indexOf(t)は、文字列 sの先頭から部分文字列 tを探し、その位置を

返すメソッドである。逆に、メソッド s.lastIndexOf(t)は、文字列 sの末尾から部

分文字列 tを探し、その位置を返すメソッドである。位置は、文字列のインデックスと同

様に 0から始まる。ともに、部分文字列 tを発見できなかった場合は、-1を返す。

>>> s = "Uzumaki Naruto"
>>> s.indexOf("Naruto")
>>> s.indexOf("Sakura")
-1
>>> s.indexOf("a")
4
>>> s.lastIndexOf("a")
9

最もよく使う利用法は、シーケンススライスと組み合わせて、文字列を分割するときで

ある。

>>> s = "Name: Uzumaki Naruto"
>>> s[.. s.indexOf(": ")]
"Name"
>>> s[s.indexOf(": ")+1 ..].trim()
"Uzumaki Naruto"

7.4.3 split(), tokenize()

メソッド s.split(t)は、文字列 sを分割文字列 tによって単純に分割するメソッ

ドである。分割された文字列は、String[]に格納される。（分割文字列 t)は、含まれ
ない。）

>>> s = "Naruto, Sakura, Sasuke"
>>> s.split(", ")
["Naruto", "Sakura", "Sasuke"]
>>> s.split(",")
["Naruto", " Sakura", " Sasuke"]

特別な用例としては、分割文字列を省略すると、文字列は 1文字ずつ split()される。

>>> s = "naruto"
>>> s.split()
["n", "a", "r", "u", "t", "o"]

もう少し意味のある単位でトークンを分割したい場合は、メソッド s.tokenize()

も利用できる。これは、文字列 sを Konohaのレキシカル構造にしたがってトークンに分
割するメソッドである。

>>> s = ’’’
print (1+1), ’hello,world’;
’’’
>>> s.tokenize()
["print", "(", "1", "+", "1", ")",

",", "’hello,world’", ";"]

7.4.4 置換: replace()

メソッド s.replace(t, u)は、文字列 s中に出現する部分文字列 tを与えられた

文字列 uによって置き換えるメソッドである。Konohaでは、文字列は不変オブジェクト
であり、破壊的操作はできないため、新しい Stringオブジェクトが生成され、メソッドの
戻り値として返される。

>>> s = "Haruno Naruto"

>>> t = s.replace("Haruno", "Uzumaki")
>>> t
"Uzumaki Naruto"
>>> s // 元の文字列は変わらない
"Haruno Naruto"

注意：replace()メソッドは、正規表現パターンによる置き換えも可能である。詳し

くは、「第 7.7節正規表現」で述べる。

7.4.5 toUpper(), toLower()

メソッド s.toUpper() は、文字列 s の英文字を英大文字に変換し、メソッド

s.toUpper()は、文字列 sの英文字を英小文字に変換するメソッドである。

>>> s = "Uzumaki Naruto"
>>> s.toUpper()
"UZUMAKI NARUTO"
>>> s.toLower()
"uzumaki naruto"

7.4.6 メソッドのパッケージ拡張

Konohaは、実行中であってもクラスに新しいメソッドを追加したり、既存の振る舞い
を変更することができる。Stringクラスも例外ではなく、パッケージのインポートによっ
て、メソッドが拡張されることがある。

>>> using japanese; // 日本語パッケージ
>>> ("なると").toKatakana() // カタカナへの変換
"ナルト"
>>> ("ナルト　").trim() // 全角空白も除去される
"ナルト"

7.5 テンプレート・フォーマッティング

フォーマッタは、オブジェクトを様々な書式で文字列化する Konoha 独自の機能であ
る。理想的には、全てのオブジェクトはクラス階層にしたがってフォーマッタが定義さ

れ、全てのフォーマッティングはオブジェクト指向モデルによってよく再利用されるべ

きである。しかし、現実問題としては、再利用性を考慮に入れるよりも、とりあえずア

ドホックなフォーマッティングが便利なケースも多い。その場合、従来のテンプレート・

フォーマッティング、つまり「テンプレートを作成しテンプレートの変数を置き換えるこ

とでフォーマッティングを行う」ことは、便利である。

Konoha では、ビルトイン関数 format() を用いることで、テンプレートとそのパラ

メータを組み合わせてフォーマットされた文字列を生成することができる。もちろん、

Konohaでは、全フォーマッタをテンプレートの書式として利用することができる。

>>> format("We are at [%s:%d]:", __file__, __line__)
"We are at [(shell):1]"

テンプレートは、テンプレート書式にしたがえば、あとは通常の文字列として扱うこと

ができる。次は、変数に格納されたテンプレートをフォーマット処理したときである。

>>> fmt = "We are at [%s:%d]:";
>>> s = format(fmt, __file__, __line__)
>>> s
"We are at [(shell):4]"

7.5.1 テンプレートの書式

テンプレート書式は、"%x{n}"スタイルで、「フォーマッタ %xに n番目のパラメータ
を適用する」と意味になる。ここでインデックスは、1から始まる。わざわざ、パラメー
タのインデックスを指定できる理由は、英語と日本語のような語順が違うテンプレートで

も、パラメータの順番を気にしなくて適用するためである。

>>> fmt = """
... print "name";
... print "name"
... """

7.5.2 フォーマッティング

テンプレートは、ビルトイン関数 format() を用いて、フォーマッティングされる。

フォーマットされた文字列は、関数の戻り値として返される。

>>> format(fmt, n, n)

フォーマッティング時のパラメータは、Konoha において数少ないコンパイル時に型
検査されないパートである。フォーマッタは、与えられたパラメータのクラスに対して、

動的にバインドされる。もし指定されたフォーマッタが定義されてなければ、自動的に

%emptyが適用され、そこには文字列が表示されない。

もし厳密な型検査を求める場合は、関数等でラッピングして用いるとよい。

7.6 インライン・フォーマッティング

Konoha では、#で始まるライン文字列リテラルに対してのみ、インデックスの代わり
に、変数やその他の式を直接埋め込むことができる、ダイレクトフォーマッティングを認

めている。

c.query
# select name, salary from persontbl
# where age > %d{age} and age < %d{age2}

7.7 正規表現

正規表現 (regular expression)は、文字列のパターンを記述する小さな言語であり、有限
オートマトンにコンパイルされ、強力なマッチングエンジンとあわせて利用されることが

多い。スクリプティング言語では、Perlを代表例として文字列処理の中心的機能として正
規表現を言語に統合している。Konohaでは、独自の「正規表現」という概念は存在しな
いが、GNU regex, Oniguruma, PCRE(Perl Compatible Regular Expression) など、既存の
正規表現ライブラリを選択して利用することができる。

7.7.1 正規表現リテラル

正規表現パターンは、Regexクラスのインスタンスとして表現される。Konohaでは、
Perl, Ruby, JavaScript などと同様に、専用の正規表現リテラルが用意されている。正規
表現リテラルは、/記号で囲まれた文字列であり、そこで表現された文字列は正規表現パ

ターンと解釈される。

pattern = /s$/

正規表現リテラルは、Regexコンストラクタで生成されるオブジェクトと全く同じであ
る。また、Regexコンストラクタは、@Const属性であるため、コンパイル時にオブジェ
クトが生成されるため、性能の面でも同じである。しかし、正規表現リテラルの方を使う

ことが圧倒的に多い。何よりもシンプルに記述できるからである。特別なオプションを指

定するときだけ、現在のところ、コンストラクタを使う必要がある。

pattern = new Regex("s$", "i"); // 英大文字/小文字を
無視..

7.7.2 正規表現エンジン

Konohaは、標準的な正規表現エンジンとして、POSIX regexライブラリを利用するこ
とが多い。現在の正規表現エンジンは、$konoha.regexで確認することができる。

>>> $konoha.regex
"GNU regex library"

正規表現エンジンは、正規表現パッケージをインポートすることで切り替えることがで

きる。主な正規表現パッケージは、以下のとおりである。

• oniguruma — kosako氏が開発し、Rubyで正式採用されている高性能な正規表現
ライブラリ

• pcre — Perl互換の正規表現ライブラリ

切り換えは、パッケージのインポートと同期している。つまり、正規表現パッケージを

インポートすれば、正規表現ドライバが入れ替えられ、それ以降の正規表現は新しいエン

ジンに切り替わる。

>>> using oniguruma;
>>> $konoha.regex
"oniguruma"
>>> pattern = /ˆ*.:*.$/ // 以後は、鬼車を利用

注意：正規表現ライブラリをインポートする以前の正規表現オブジェクトは更新が反映

されない。

また、正規表現ライブラリごとの異なる機能を明示的に使い分けたいときは、正規表現

リテラルの代わりに、’regex:’、’pcre:’、’onig:’のように、それぞれのライブラ

リを識別する意味タグを用いることで、正規表現を用いることができる。

>>> using oniguruma;
>>> s =˜ ’onig:ˆ*.:*.$’; 　　　　// 鬼車でマッチング
>>> s =˜ ’regex:ˆ*.:*.$’; // regexでマッチング

7.7.3 正規表現と演算子

Konoha は、Perl 由来の正規表現マッチング演算子をサポートしている。しかし、
JavaScriptもサポートしていないようだから、この演算子は中止しようかと悩んでいる。

>>> s = "name: naruto";

>>> s =˜ /ˆ*.: *.$/
true

7.7.4 正規表現とメソッド

Stringクラスは、いくつかのメソッドを正規表現パターンに対応させている。
s.serach(p)は、文字列 sの先頭から、正規表現パターン pでマッチング検索を行

い、最初にマッチした文字列インデックスを返す。indexOf()の正規表現版である。

>>> text.search(/ha/);

s.match(p)は、文字列 sに対して、正規表現パターン pでマッチングを行い、その

マッチング結果の文字列を返す。複数個マッチした場合は、全て返す。

>>> s = "1 + 2 = 3";
>>> s.match(/\d/);
["1", "2", "3"]

また、正規表現パターンにおいてグルーピングを用いれば、マッチングした文字列に続

いて、グルーピングされた部分文字列も得ることできる。

>>> url = "http://www.website.org/konoha";
>>> url.match(/(\w+):\/\/([\w.]+)\/(\S*)/);
["http://www.website.org/konoha",

"http", "www.website.org", "konoha"]

s.replace(p, fmt)は、文字列 sに対して、正規表現パターン pでマッチングを

行い、その結果をフォーマッティング fmtによって置き換える演算子である。

>>> s = "1 + 2 = 3";
>>> s.replace(/(\d)/, "%s{1}.0");
"1.0 + 2.0 = 3.0"

7.7.5 正規表現とマッピング**

最近の Perl 互換の正規表現では、名前付きのグルーピングを提供している。文字列を
正規表現からパースして、直接オブジェクトへマッピングする機能を計画している。

>>> Class Url {
... String site;
... String path;
... }

>>> url = "http://www.website.org/konoha";
>>> pattern = /(\w+):\/\/(?site:[\w.]+)\/(?path:\S*)/;
>>> u = (Url with pattern)url;　　　// 何言語だ？
>>> u
{site: "www.website.org", path: "/kohoha"}

第 8章

配列とリスト

配列 (array)は、複数個のオブジェクトを扱う最も基本的なデータ構造である。Konoha
では、配列は Arrayクラスとして実現されている。名前は、シンプルに Arrayであるが、
Javaの ArrayListと同様に、可変長の成長する配列（Growing Array）であり、リストとし
ても利用できる。

8.1 配列と型

Konohaは、Array型と C[]型の、２種類のスタイルの配列が存在するようにみえる。

Array型は、任意の型の要素をもつのに対し、C[]はある特定の C型の要素のみもつ。し
かし、両者は、ともに Arrayのいわゆる総称型の別名で、つまるところ、Array<Any>
と Array<C>である。基本的な性質は大きく変わらない。

本書では、Konohaスタイルで型を書くが、実際のソースコードでは配列スタイル、総
称型スタイルのどれを利用しても構わない。

Konoha 配列 総称型
Array Any[] Array<Any>
int[] int[] Array<int>
String[] String[] Array<String>
C[] C[] Array<C>

注意: int[]は、Array<int>と同じであるとなっているが、あくまでも見かけ上の
話である。Konoha内部では、より正確に言えば、unboxされた形で整数要素を格納してい
る。同様に、float[]も unboxされた配列である。ちなみに、boolean[]は、unbox
された配列ではない。

8.1.1 Any[]と Object[]の違い

Object は、全てのオブジェクトの上位クラスであるため、Object[] は Any[]、つ

まり Arrayと同様に任意の型のオブジェクトを要素に持つことができる。ただし、「第 ??
章　ダイナミック言語の特徴」で述べるとおり、Objectと Anyは、キャストの取扱いが
異なるため、両者には注意すべき違いが生じる。

まず、Object[]の方であるが、基本的に Javaの配列と振る舞いは同じである。ただ
し、要素の型は、コンパイルした時点では Object型であり、次の例のとおり、実行時の
要素が String型であっても、明示的なダウンキャストが必要となる。

>>> String[] o = ["naruto", "sakura"];
>>> typeof(o[0]) // 型は Object
Object
>>> String s = (String)o[0]; // ダウンキャスト
>>> s
"naruto"

これに対し、Any型は、実行時に動的な型検証が入るため、静的なキャストは不要にな

る。こちらの方が便利な反面、暗示的に型キャストをしているため気づかないで実行効率

が落ちることがある。

>>> Any[] a = ["naruto", "sakura"];
>>> typeof(a[0]) // 型は動的
Any
>>> String s = a[0]; // キャストは不要
>>> s
"naruto"

8.1.2 null要素

Konohaの配列は、null値を配列要素に持つことができない。ただし、Array型 (Any[])
は、そもそも Any型が nullを認めるため、nullを要素にもつことができる。

>>> String[] s = ["naruto"];
>>> s[0] = null;
** Null!!

8.1.3 バイト列 byte[]

Konoha の整数型は、64 ビット長の Int 型に単一化されているため、バイト (8 ビット
整数) を表す型はない。多少、メモリが無駄かも知れないが、int でバイトを表すことに
なる。しかし、バイト列（バイナリデータ）となると、int[]を使うのでは多少の無駄

では済まない。そこで、バイト列を扱う専用のクラスとして、Bytesが用意されている。
byte[]は、ごたぶんに漏れず、Bytesの別名である。

byte[] buf = new byte[4096];

バイト列 byte[]は、配列とほぼ同じ性質を持ち、本章で取り上げるメソッドや演算

子をほぼそのまま利用することができる。バイト列特有の性質のみ、「第 8.6節バイト列」
で取り上げる。

8.1.4 共変性**

8.2 配列の生成

配列は、配列リテラルで配列をデータとして与えるときと、コンストラクタを用いて配

列を作る場合の 2種類の方法がある。

8.2.1 配列リテラル

配列リテラルは、Pythonや JavaScriptなどで採用されている []コンストラクタを用い

る。重要な相違点は、リストの要素によって配列の型が推論される点である。

配列リテラルの型推論リールは、いたって単純である。まず、最初の要素の型 T を配列

型 T []の候補として、残りの要素がすべて T か、そのサブタイプであるとき、T[]と推

論する。そうでないときは、Arrayクラスとなる。

[] // 空の Array
[1,2,3] // Int[] の配列
["Naruto", "Sakura"] // String[]の配列
["Naruto", 9] // Array

8.2.2 Arrayコンストラクタ

Arrayクラスのコンストラクタは、2種類存在する。ひとつは、Arrayをリストとして
利用するため、空の Arrayを作る方法である。

>>> a = new Array(10); // 空の Arrayを生成
>>> a
[]

もうひとつは、Java言語と同様に、配列の大きさをあらかじめ当てはめて宣言する方法
である。

>>> a = new Any[4]; // 要素数 n個の配列
>>> a
[null, null, null, null]

8.2.3 C[]コンストラクタ

配列のコンストラクタは、Java言語と同様に、配列の大きさをあらかじめ当てはめて宣
言する方法である。

>>> ss = new String[4]; // 要素数 4個の配列 String[]
>>> ss
["", "", "", ""]

生成された配列の要素は、要素型のデフォルト値 (default(C))で満たされる。(Java
と異なり、nullではない。)デフォルト値が不都合な場合は、要素数を 0にして空の配列
を生成したのち、追加することになる。

>>> ss = new String[0]; // 空の String[]
>>> ss
[]

8.3 配列と演算子

配列は、0個以上の要素が並んだ典型的なシーケンスである。Konohaでは、シーケン
ス sは、|s|演算子を用いることで、その長さを得ることができる。配列の場合は、要素
の数である。

>>> a = [0, 1, 2, 3]
>>> |a|
4

配列 aのインデックスは、C/C++, Java言語と同様に 0から始まり、|a| - 1である。

その範囲を超えると、OutOfIndex!!例外が通知される。

>>> a = [0, 1, 2, 3]
>>> a[0]
0
>>> a[|a|-1]
3
>>> a[4]
OutOfIndex!!

配列は、基本的に変更可能である。正しいインデックスを指定すれば配列の値を変更で

きる。

>>> a = [0, 1, 2, 3]

>>> a[1] = 10 // a[1]を変更
>>> a
[0, 10, 2, 3]

注意：初期の Konohaでは、python風に最後尾からインデックスを数える −n表記が利

用できた。このインデックスは便利であったが、C/C++, Javaなどの既存の配列処理との
互換性や最適化処理の困難さを理由に廃止となった。

8.3.1 スライシング

配列 a から部分配列を取り出すときは、スライス演算子 a[m..n] を用いることがで

きる。mと nは、省略可能で、省略した場合は、それぞれ 0と最後の添字 (|a|-1)と解
釈される。

>>> a = [0, 1, 2, 3, 4, 5, 6, 7, 8]
>>> a[2..] // a[2]から
[2, 3, 4, 5, 6, 7, 8]
>>> a[2..6] // a[2]から a[6]まで
[2, 3, 4, 5, 6]
>>> a[..<6] // a[2]から a[5]まで
[2, 3, 4, 5]
>>> a[2..+6] // a[2]から 6個
[2, 3, 4, 5, 6, 7]
>>> a[..6] // a[6]まで
[0, 1, 2, 3, 4, 5, 6]
>>> a[..<6] // a[6]まで（含まない)
[0, 1, 2, 3, 4, 5]
>>> a[..+6] // 最初の 6個
[0, 1, 2, 3, 4, 5, 6]

8.3.2 リストとセット in?

8.4 多次元配列

Konohaの配列は、基本的に 1次元配列である。n次元配列を使うというのは、単純に、
C 言語同様に適切に次元を計算してからインデックス操作を行えばよい。例えば、それ
ぞれの次元の大きさが X, Y, Zの 3次元配列を使いたい場合は、次のようにインデックス
(index)を計算できる。

int[] a = new int[X * Y * Z];
index = x + X * y + (X * Y) * z;
a[index]

当初、多次元配列に関しては、これで十分だと考えていた。それは、Konoha言語の開
発を含め、過去のプログラミング実践で多次元配列がなくて困った経験をほとんどしな

かったためである。しかし、これもクラスルームから来た要求であるが、初期のプログラ

ミング練習課題は、オセロやら五目並べやら、やたらと 2次元配列を使いたくなることが
多い。学生から、「2次元配列は使えないのですか？」と質問される度に、上記の式を書く
より、シンタックスシュガーでさっさと対応すべきではないか？、ということになった。

そういうわけで、多次元配列は、型としては 1次元配列のときと同じである。ただし、
コンストラクタで次元を設定すると、次元情報はメタデータとして記録され、このメタ情

報を用いて多次元からインデックスをを再計算している。確かに、少しすっきりする。

int[] a = new int[X, Y, Z];
a[x, y, z]

注意：次元をメタ情報として管理しているため、次元数に上限が生じる。Konohaでは、
最もよく使うと考えられる 2次元、もしくは 3次元配列のみサポートされている。4次元
配列が必要なアプリケーションを書いておられる方は、是非、連絡して欲しい。

次は、2次元配列の利用例である。クラスルームで使う程度であれば、ふつうの 2次元
配列として利用することができる。

>>> a = new Int[8, 8]; // 2次元配列
>>> a[1, 7] = 1
>>> a[1, 7]
1
>>> a[8, 1] // 次元単位でチェック
** OutOfIndex!!

多次元配列は、1次元配列なので、1次元配列として操作することもできる。多次元配
列としての型安全性は備えていない。

>>> a = new Int[8, 8]; // 2次元配列
>>> typeof(a)
Int[]
>>> |a| // 大きさは、1次元
64
>>> a.add(100) // 追加もできる
>>> |a|
65
>>> a[64] // 1次元配列の操作も可
100

8.5 配列とメソッド

Konohaは、Javaの Collectionフレームワークの標準的なメソッドをベースにして Array
関係のメソッドを定義している。

8.5.1 リスト: add(), insert(), remove()

Konohaの配列は、全て可変長であり、リストのように要素数のわからないデータを実
行時に必要に応じて伸縮させながら利用できる。

メソッド add(v)は、最もよく使われるメソッドである。配列の最後尾に新しい要素

vを追加する。演算子 <<と同じ働きをする。

>>> a = [0, 1, 2]
>>> a.add(100)　　　　　　　　　　// 追加
>>> a
[0, 1, 2, 100]
>>> a.add(30000) // 追加
>>> a
[0, 1, 2, 100, 300000]

メソッド insert(n, v) は、n 番目に要素 v を挿入するときに利用し、メソッド

remove(n) は、n 番目からを取り除くとき利用する。どちらの場合は、配列の大きさ

は、それぞれ増減する。

>>> a = [0, 1, 2]
>>> a.insert(1, 100)　　　　　　　// 挿入
>>> a
[0, 100, 1, 2]
>>> a.remove(2) // 削除
>>> a
[0, 100, 2]

Konoha の配列は、リストのように利用できるが、双方向リンクリスト構造ではない
ため、insert(), remove() ともに配列要素をコピーするだけのコストがかかる。

add()は、十分に高速に要素を追加することができる。

8.5.2 キューとスタック: first(), pop()

Konohaの配列は、キュー (FIFO)やスタック (FILO)として利用することができる。

>>> queue = []
>>> queue.push(1) // キューに追加 add()と同
じ

>>> v = queue.first() // キューから取り出す

>>> stack = []
>>> stack.push(1) // スタックに追加 add()と
同じ
>>> v = stack.pop() // スタックから取り出す

注意：配列は、スタックとして利用するとき、配列の大きさに関わらず、十分に高速で

ある。しかし、キュー操作の場合は、first()のとき配列要素のコピーが発生するため、

配列の大きさに応じてコストがかかる。

8.5.3 reverse(), shuffle()

Konohaは、...

8.5.4 整列：sort()

Konohaは、高速なソートアルゴリズムによって配列の要素の並びかえることができる。
大小の比較は、<演算子によって行われる。

>>> a = [0, 2, 1]
>>> a.sort() 　　　　　　　// ソート
>>> a
[0, 1, 2]

注意：sort()は、いわゆる昇順ソートである。降順ソートのためのメソッドは用意さ

れておらず、sort()と reverse()を組み合わせて用いると、降順になる。

8.5.5 サーチ: indexOf(), lastIndexOf(), binarySearch()

サーチは、配列中から指定された要素を探すことである。配列は、indexOf()は、先

頭から探して最初に見つかったインデックスを返す。lastIndexOf()は、逆に最後尾

から検索して最初に見つかった箇所のインデックスを返す。ともに発見できなかった場合

は、-1を返す。

>>> a = [1, 2, 3, 2, 1]
>>> a.indexOf(2) 　　　　　　　// 先頭から検索
1
>>> a.lastIndexOf(2) 　　　　　// 後尾から検索

3
>>> a.indexOf(0) // 見つからない
-1

indexOf()と lastIndexOf()は、どのような配列に対しても検索が可能である。

しかし、単純に順番に検索するため、検索時間は配列の大きさに比例し、効率はあまりよい

といえない。もし配列が sort() で整列された状態であるときは、binarySearch()

を用いると、バイナリーサーチアルゴリズムにより効率よく検索することができる。

>>> a = [1, 3, 2, 4, 0]
>>> a.sort()
>>> a
[0, 1, 2, 3, 4]
>>> a.binarySearch(3) // バイナリサーチ
3
>>> a.binarySearch(5) // 見つからない
-1

8.6 バイト配列: byte[]

バイト配列は、バイナリデータを扱うための専用の配列であり、ヒープ上のメモリを確

保し、それをダイレクトに扱うことができる。

>>> buffer = new byte[4096];
>>> typeof(buffer)
Bytes

sort()や shuffle()など、バイナリ操作にあまり関係ないメソッドをサポートし

ていないが、それ以外は配列と同じように操作することができる。

第 9章

関数

Konohaにおける関数は、Scriptクラスのメソッドである。しかし、関数のアイディア
はオブジェクト指向プログラミングより古く、クラス概念を導入しなくても利用できる伝

統的なモジュールであるため、関数風に利用できる、つまりレシーバーオブジェクトを省

略できるメソッドを「関数」と呼んでいる。

9.1 関数の定義

関数 (function)は、プログラムの一部をモジュール化するための代表的な手段である。
名前が示すとおり、数学における関数 y = f(x)をメタファ*1として、引数 xに対して結

果 y が得られるという形式でモジュール化を行う。

次は、f(x) = x + 1という式を関数でモジュール化した例である。一旦、関数を定義し
てしまえば、関数コール (呼び出し) によって、モジュール化されたプログラムを利用す
ることができる。関数は、引数によってモジュールのパラメータ化が可能であり、引数に

よって異なるモジュール実行の結果が得られる。

>>> int f(int x) { return x + 1; }
>>> y = f(1)
>>> y
2
>>> f(2)
3

プログラマは、既存の関数を利用するだけなく、新しい関数を自由に定義することがで

きる。関数は、Konoha名前規則に従い、英小文字で始まる名前を持ち、0個以上の引数
(arguments) をとり、モジュール本体のプログラムが続く。モジュールの処理が終れば、
その評価結果を返す。これを戻り値 (return value)、もしくは返り値と呼ぶが、特定の値を

*1 Konohaは、関数型プログラミング言語でないため、純粋な数学的性質は保証されないため、メタファと
している。

戻さなくても構わない。

9.1.1 関数宣言と型

プログラミング言語における関数は、引数や戻り値において「型」が必要な点が特徴で

ある。Konohaは、C/C++, Java言語などと同様に、関数宣言にはこれらの型を明示的に宣
言しなければならない。次は、C型、D型、E型の引数をとり、T型の値を返す関数宣言
である。

T f(C x, D y, E z) { ... }

これは、変数宣言では型推論を前提として省略を認めていることに対し、関数定義では

極めて保守的なスタンスをとっているように思われるかも知れない。しかし、この辺りは

プログラミング言語設計の技術論だけで決められない部分でもある。特に、関数（及びメ

ソッド）は、異なる開発主体が開発するソフトウェア境界のインターフェースとなる。そ

して、現在のソフトウェア開発ではインターフェースを明示的に仕様化することが望まし

いとされ、更に型検査などで機械的に検査されることがもっと望ましいとされている。そ

ういうわけで、Konohaでは、関数定義は必ず型を明確に宣言する方針をとっている。
一方、いろいろな誘惑もある。JavaScriptとの互換性、クロージャなどの記述性のため、
ひょっとしたら関数の引数、もしくは戻り値に対し、型推論が導入されることは十分に考

えられる。

function f(x, y, z) {...}

9.1.2 引数と参照渡し

関数の引数は、モジュール間のデータをやり取りするインターフェースになる。プログ

ラミング言語の教科書を読めば、値渡し (call by value)、参照渡し (call by reference)、名
前渡し (call by name)の３種類存在することになっているが、

Konohaは、結論から言えば、全て参照渡し (call by reference)である。理由は、”Every-
thing is an Object”の世界であり、関数コールの度にオブジェクトをコピーするのはコスト
が大きすぎるためである。オブジェクト自体の代わりに、オブジェクトの参照、つまりオ

ブジェクトのポインタを渡すことで、効率よい関数コールを実現している。

>>> void f(String t) { print %p(t); }
>>> s="a"
>>> print %p(s)
[(shell):3] "0x6cc00" // s と
>>> f(s)
[(shell):4] "0x6cc00" // t は同じポインタ

>>>

参照渡しで注意すべき点は、関数内部でオブジェクトの値を変更すると、関数を呼び出

した外部でもその変更が継続することである。(関数の外と内で、同じオブジェクトを参
照しているから当たり前といえば、当たり前である。)

>>> void g(Array a) {
... a << 1;
... }
>>> Array list = [];
>>> list
[]
>>> g(list)
>>> list // 関数内のリスト操作が残る
[1]

注意：Int, Float, Stringオブジェクトは、不変オブジェクトなので、関数内で変更して
も、そのとき実際はオブジェクトのコピーが行われている。そのため、関数内部の変更に

よって外部のオブジェクトまで変更されることはない。

9.1.3 関数内同一スコープ

Konohaは、C/C++や Javaとは大きく異なり、ブロックレベルの変数スコープをサポー
トしていない。関数内で宣言された全ての変数は、どこで宣言されても関数内で利用でき

る。ただし、変数宣言より先に変数を使うことはできない。

例えば、次のコードにおいて、変数 k, jは、スコープ内として扱われる。

void test(int n) {
int i = 0;
if(n > 10) {

int j = 0;
for(int k = 0; k < n; k++) {

print k;
}
print k; // スコー

プ外ではない
}
print j; // スコープ外

ではない
}

次のような例外処理のコードを書くときも都合がよい。

try {
InputStream in = new InputStream("file.txt");
for(String line from in) {

print line;

}
}
catch(IO!! e) {

print format("Error at %d", in.line);
}
finally {

in.close();
}

また、Konohaは同一スコープ内であっても同じ型であれば、何回、型宣言を行っても
エラーとならない。そのため、コピー＆ペーストする場合も支障とならない。

void test(int n) {
int i = 0;
if(n > 10) {

int j = 0;
for(int k = 0; k < n; k++) {

print k;
}
print k; // スコー

プ外ではない
}
print j; // スコープ外

ではない
}

9.1.4 可変長パラメータ**

Konohaは、可変長変数を採用することができる。

9.1.5 パラメータ初期値*

9.1.6 戻り値*

9.2 関数コール

9.3 ビルトイン関数

関数は、通常、プログラム実行時に評価される。しかし、いくつかのプログラミング言

語機能は、コンパイル時に利用したいことがあり、コンパイラと密接に結びついた機能を

ビルトイン関数として提供している。

9.3.1 スタティックな型付け typeof()

ビルトイン関数 typeof(expr)は、プログラム中で、与えられた式 exprのスタティッ
クな型を調べるビルトイン関数である。

>>> a = 1
>>> typeof(a)
Int
>>> typeof("hello,world")
String

これに対し、(expr).classは、実行時の exprがもつクラス（型）を調べるメソッド
である。typeof(expr)と instanceofの関係が成り立つが、必ずしも一致するとは限らない。
また、型をもたないトークン（例えば予約語など)を評価した場合、void型が返される。
typeof()関数の評価は、コンパイラの型検査時に行われる。つまり、あまりおすす

めではないが、次のように型名の代わりに型宣言にも利用できる。

int f(int n) {
typeof(n) m = n + 1;
return m;

}

9.3.2 識別子の存在 defined()

ビルトイン関数 defined(name)は、識別子 nameが定義されているかどうか調べる
関数である。

>>> defined(Class)
true
>>> defined(System.LINUX)
false
>>> defined(a)
false
>>> a = 1
>>> defined(a)
true

defined()関数の評価も、コンパイラの型検査時に行われ、必ず論理値 trueもしく
は falseに置き換えられている。

if(!defined(System.LINUX)) {
print os.uname;

}



82 第 9章 関数

9.3.3 デフォルト値をえる default()

デフォルト関数 default(expr) は、プログラム中で、与えられた式 expr の型から
そのデフォルト値をえる関数である。

>>> default(String)
""
>>> default(%s(1))
""
>>> default(Script) // Script は実行依存
main.Script

多くのデフォルト値は、スタティックに解決される。しかし、いくつかのクラスは、実行

コンテクストに依存する。そのため、デフォルト値も存在する。そのため、default()

は typeof()と同様にスタティックに型まで決定し、一部のデフォルト値は実行時に得

ている。

9.3.4 デリゲートの生成 delegate()

ビルトイン関数 delegate()は、デリゲートクロージャを生成する関数である。デリ

ゲートの詳細は、「第??節デリゲート」で述べる。
デリゲート delegate(o, m)は、閉包するオブジェクト oと呼び出すメソッド名 g

をセットにしてクロージャを生成する。

>>> String s = "name";
>>> f = delegate(s, split);
>>> f() // s.split() が呼ばれる
["n", "a", "m", "e"]

もしくは、delegate(f)のように関数名 fのみを引数として与え、クロージャを生

成することもできる。

>>> g = delegate(fibo);
>>> g(10) // Script.fibo(10) が
呼ばれる
55

9.3.5 フォーマッティング format()

format() は、「第??節 テンプレート・フォーマッティング」のとおり、テンプレー
ト・フォーマッティングのためのビルトイン関数である。この関数がビルトイン関数であ

る理由は、もし可能であれば、テンプレートのインライン展開を行い、フォーマッティン

グの性能を向上させるためである。

9.3.6 likely()/unlikely()*

likely()/unlikely() は、論理値 (true/false) を受け取り、その値をそのまま返す
ビルトイン関数である。ほとんどのプログラマには、この関数の存在意義からして疑わし

いものかも知れないが、Linuxカーネル開発者など一部からは非常に好まれている。少な
くとも、ソースコードが読みやすくなるという効果は認められるようである。

>>> if(unlikely(a == b)) {
... // あまり起こらないケース
... }
>>>

現在、Konohaでは、クラスルーム利用において、コンピュータアーキテクチャと結び
つけて、プログラミングの深淵（の一部）を説明するときに役立つように導入している。

将来、優れた Konohaコンパイラが登場したとき、likely()/unlikely()をひょっ
としたら正しく解釈して、分岐予測の最適化を行ってくれるかも知れない。現在は、無

害で副作用のないビルトイン関数であるが、もし使うのなら正しく使うことをオススメ

する。

クラスとオブジェクト

Konoha のオブジェクト指向プログラミングは、名前ベースのクラス (nominal class)、
単一継承 (single-inheritance)など、主に Java言語のそれから影響を受けている。

10.1 class 宣言

新しいクラスは、class文を用いて宣言することで定義できる。class文の最も簡単なシ
ンタックスは、次のとおりである。クラス継承を含めたシンタックスは、「第??節クラス
継承」で扱う。

class ClassName {

　　 ClassMembers //クラスのメンバー
}

クラスのメンバーには、フィールド変数 (field variable)、メソッド (method)、コンスト
ラクタ (constructor)が含まれる。

• フィールド変数は、SmallTalkではインスタンス変数、C++ではメンバ変数、Java
ではフィールドとそろぞれ異なる呼称をもつが、オブジェクトの内部状態を表す変

数である。Javaではフィールド変数と呼ばないのが正しいが、Konohaでは変数の
種類をその位置によって区別する統一呼称を用いているので、「フィールド変数」

と呼んでいる。

• メソッドは、SmallTalkではメッセージ、C++ではメンバ関数とも呼ばれるが、オ
ブジェクトの内部状態を操作する手続きである。メソッドは、関数のフォームをと

り、パラメータ（メソッドにおける引数のこと）を受けて処理を行い、その結果を

戻り値として返す。Konohaでは、オブジェクトへの操作が明確なときはメソッド/
パラメータ、そうでないときは手続き型言語風に関数/引数と呼び分けている。ど
ちらも同じである。

• コンストラクタは、新しくオブジェクトを生成するときに、new演算子を用いて呼

び出される特別なメソッドである。なお、Konohaは、ガベージコレクションの機
構を備えているため、明示的なオブジェクト破壊を行う必要なない。

さて、次は Konohaにおけるクラス定義の例である。フィールド変数、コンストラクタ、
メソッドの定義をもっている。フィールド変数は、classブロックの一番先頭で宣言する
ことが（よみやすいため）一般的であるが、定義する順番は自由に変えても構わない。

class Person {
String _name; // フィールド変数
int age;

// コンストラクタ
Person(String name, int age) {

_name = name;
_age = age;

}
String getName () { // メソッド

return _name;
}

}

Konohaは、スクリプティング言語の柔軟さを提供するため、メソッドやコンストラク
タを既存のクラスに追加するを認めている。しかし、フィールド変数のみは、class ブ
ロックであらかじめ定義しなければならない。

10.1.1 クラスの種類

Konohaは、いくつかの特性の異なったクラスを分類している。これらは、class 文の
前に、アノテーションを与えることで宣言することができる。

アノテーション 説明
@Private 名前空間からみることはできない。
@Final これ以上、継承することはできない
@Singleton Singleton 関数
@Interface インターフェースとして利用することが可能
@Release デバッグの完了したクラス

10.1.2 Glueクラス

10.2 フィールド変数

フィールド変数は、アンダースコア (_)で始まる変数名をもつ。これは、Javaプログラ
マがよく使っているローカル変数とフィールドを区別する慣習に由来するが、フィールド

変数宣言のとき、アンダースコアを付け忘れなくても、コンパイラは自動的にフィールド

変数にアンダースコアを付加する。

class Person {
String _name;
int age; // もし _ を付けなくても
Person(String name, int age) {

_name = name;
_age = age; // _ は、フィールド変数へのアクセス

に必要
}

}

重要な留意点は、フィールド変数宣言のとき、アンダースコアを付けないことは、自動

的な getter/setterメソッドの生成を意味する。これは、次節で述べるとおり、フィールド
変数の public宣言に等しい。

class Person {
...
int age;
int getAge() { return _age;} // 自動生成
void setAge(int age) { _age = age; } // 自動生成

}

10.2.1 フィールドアクセサ

Konohaは、全てのフィールド変数はいわゆる privateである。オブジェクト外部から
直接参照はできず、全て getter/setterメソッドを用いて行う。

>>> p = new Person("naruto", 17);
>>> p.getName();
"naruto"
>>> p.setAge(18);
>>> p.getAge();
18

フィールドへのアクセスに getter/setterメソッドを多用するのは、Javaが産み出したト
レンドである。しかし、これは何かと不便に感じる人も多い。そこで、Konohaは、フィー
ルドアクセサ (. 演算子) を、getter/setter を呼び出すための省略形 (シンタックスシュ
ガー）として再利用している。

>>> p = new Person("naruto", 17);
>>> p.name; // 実は p.getName()
"naruto"
>>> p.age = 18; // 実は p.setAge(18)
>>> p.age;
18

今までどおり、フィールドアクセスをしているように見えて、実は全て getter/setterメ
ソッドによってふるまいを変更することができる。

10.2.2 バーチャルフィールド

バーチャルフィールドとは、フィールド変数の実体がなくても、getter/setterを通して、
あたかもフィールドが存在するようにふるまうことである。

class Person {
...
String name;
String getFirstName() {

return _name.split(" ")[0];
}

}

>>> p = new Person("naruto uzumaki", 17);
>>> p.name // 実は p.getName()
"naruto uzumaki"
>>> p.firstName // 実は p.getFirstName()
"naruto"

Konohaでは、そのオブジェクトのクラス Classインタフェースを通して、フィールド
変数、（バーチャル）フィールドのリストを得ることができる。

>>> p = new Person("naruto", 17);
>>> p.class.fiedls("variable")
["_name", "_age"]
>>> p.class.fields("getter")
["name", "age", "firstName", "lastName"]
>>> p.class.fields("setter")
["name", "age"]

10.2.3 this 参照

キーワード thisは、自オブジェクト参照への参照である。thisを用いたフィールド

アクセサを用いると、フィールド変数を直接参照する代わりに、getter/setter を経由して
アクセスすることになる。

...
int age;
boolean isDrinkable() {

return (this.age => 20);
}

フィールドアクセサによる特別な結果を期待する場合でもない限り、_age で直接

フィールド変数を参照した方が高速である。次のように、getter/setter メソッド内で this
参照を用いると、予期しない依存関係から無限ループに陥る危険性もある。

int _age; // やってはいけない
int getAge() { return this.age;}
void setAge(int age) { this.age = age; }

Konohaコンパイラは、getter/setter内では、this参照を使うとエラーになる。

10.3 メソッド

メソッドは、オブジェクトに対してメッセージを送受信する手段である。0個以上のパ
ラメータを受け取り、0 もしくは 1 つの戻り値を返すことができる。次の isChild()

は、0個のパラメータ、Boolean型の戻り値をもったメソッドの例である。

class Person {
String name;
int age;

...
boolean isChild () {

return (_age < 21);
}

}

10.3.1 メソッド追加

Java 言語では、class 文の中でのみメソッドを定義することができました。Konoha で
は、動的言語の柔軟さを実現するため、既存のクラスに対して、スクリプト中どこでも新

たなメソッドを追加定義することができます。次の Perso.isChild() 関数は、class Person
内で定義された isChild()と同じ定義になります。

boolean Person.isChild () {
return (_age < 21);

}

メソッド追加は、これからインスタンス化されるオブジェクトのみでなく、（追加した

時点で）既にインスタンス化されたオブジェクトにも影響が及びます。

>>> Person p = new Person("naruto", 17);
>>> boolean Person.isChild () {
... return (_age < 21);
... }

>>> p.isChild()
true

10.3.2 抽象メソッド

抽象メソッドは、メソッドのインターフェースのみ定義され、実装パートが定義されて

いないメソッドである。Konohaでは、メソッドにつづくブロックが存在しなければ、抽
象メソッドと解釈される。また、メソッドのコンパイルに失敗したとき、自動的に抽象メ

ソッドとして扱われる。

class Person {
String name;
int age;

...
boolean hasFriend(); // 抽象メソッド

}

抽象メソッドの本来の用途は、抽象クラスやインターフェースなど、クラス設計におい

てポリモーフィズムを実現する手段としての利用である。厳密なスタティック言語では、

全ての抽象クラスが実装されていない限り、クラスのインスタンス化はできない。しか

し、Konohaは、抽象メソッドが含まれているクラスであっても、インスタンス化を認め
ている。これは、ラピッドプロトタイピングにおいて、単純に抽象メソッドを未実装なメ

ソッドとして扱うためである。

抽象メソッドを実行すると、実行時例外となる。

>>> p = new Person("naruto", 17);
>>> p.hasFriend()
AbstractMethod!!: Person.hasFriend()

我々は、実行前にスタティックな抽象メソッドを検査する機構を検討している。現在の

ところ、実行時に isAbstract()によって検査する方法しかない。

>>> String.isAbstract()
false
>>> Person.isAbstarct()
true

10.3.3 パラメータの初期値

メソッドの多重定義 (overloding)は、異なる型のパラメータを同じ名前で定義すること
である。オブジェクトプログラミング言語の標準的な機能として、C++や Java言語で広

く採用されているが、Konohaはメソッド再定義をサポートしていない。その代替機能と
して、パラメータの初期値を設定できる。

class {
...
void Person.say(String msg="hello") {

OUT << _name << " says " + msg << EOL;
}

}

初期値が設定されたパラメータは、Nullable型となる。したがって、パラメータを省略
すると、自動的に nullが与えられたものと解釈され、設定された初期値が渡される。

>>> Person.says;
void main.Person.says(String? msg);
>>> p.say("aloha");
naruto says aloha
>>> p.say();　　 // 省
略, p.say(null) と同義
naruto says aloha

Konohaは、メソッドの多重定義をサポートしていない理由は、パラメータ初期値を活
用することで、多くの場合のメソッド多重定義のバリエーションをカバーできるからであ

る。あわせて、マップキャスト機能や Any型パラメータを活用することで、いくつもメ
ソッドを多重定義することなしに、同等なプログラミングが可能となる。

10.3.4 メソッドのバージョニング

Konoha は、複数のメソッド実装をバージョン管理する機構を備えている。まず、メ
ソッドのバージョニングは、オリジナルなメソッドに対し、パラメータの型と戻り値の型

はそのまま同じで、メソッド名にバージョンタグ (:tag)を付けることで作成できる。
次は、:JAタグを付加して定義した says()メソッドの日本語版である。

void Person.says:JA(String msg = "こんにちは") {
OUT << _naame << "は、"　<< msg << "と言っ

た．" << EOL;
}

注意：（まだ厳密に決まった仕様ではないが、）英大文字で始まるタグは、言語ロケール

用に予約されている。小文字で始まるタグは、ユーザが自由に利用して構わない。

バージョニングされたメソッドは、呼び出すときに明示的にバージョンタグを付けるこ

とで呼び出すことができる。

>>> p.says();



92 第 10章 クラスとオブジェクト

naruto says hello
>>> p.says:JA();　　 // 日本語版の利用
naruto はこんにちはと言った.

将来の拡張計画

将来の Konohaでは、型による自動的なバージョン切り換えも導入を予定している。こ
の機構には、セマンティックプログラミングの意味タグ型を流用する予定であるが、セマ

ンティックプログラミングとの一貫性に関して検討を必要としている。

次は、Personクラスに対し、意味タグを追加して、意味タグ型 Person:JA型の例であ
る。意味タグ型は、同じタグが付けられたメソッドを優先して呼び出す出されることに

なる。

>>> Person:JA p = new Person("なると", 17);
>>> p.says(); // says:JA()を利用
する
naruto はこんにちはと言った.

10.4 クラス関数

クラス関数は、クラス名を直接レシーバにして利用できる特別なメソッドである。通

常、メソッドは、クラスのオブジェクトを操作するための手段である。したがって、メ

ソッドを利用するとき、クラスがインスタンス化されている必要がある。しかし、特定の

インスタンスから独立しており、インスタンスを特定できない、もしくはする必要がない

場合がある。

乱数生成メソッド random()は、特定の数値（インスタンス）の操作というより、そ

れに独立したユーティリティ機能となる。ただし、Intや Floatクラス程度に操作を区別
する必要もある。Konohaでは、これらは、クラス関数としてそれぞれ定義されて提供さ
れている。

>>> Int.random()
143980198322
>>> Float.random()
0.39810

クラス関数は、メソッド定義のとき、@Staticアノテーションを付けることで定義す

ることができる。

@Static
int Int.max(int a, int b) {

return (a > b) ? : a : b;

}

@Staticアノテーションは、Javaのスタティックメソッドに由来している。事実、見
かけ上、クラス関数はスタティックメソッドと同様に利用することができる。しかし、実

際はクラスのデフォルト値をレシーバとしてメソッドを呼んでいる。thisキーワードを用
いれば、自オブジェクトとして参照もできる。

>>> Int.max(1, 2)
2
>>> (default(Int)).max(1, 2) // 同じ
2

また、シングルトンクラスは、@Staticアノテーションがなくても、自動的にクラス

関数として解釈される。そのため、クラス関数としてコールしてもよいし、メソッドとし

てコールしても構わない。

>>> System.exit() // System はシングルトン
>>> os = default(System)
>>> os.exit() // 同じ

Konohaは、シンプルな言語構造を実現するため、Java風のスタティックフィールドを
サポートしていない。クラスで共有したい変数は、単にスクリプト変数を用いることがで

きる。

10.5 コンストラクタ

10.6 オペレータとメソッド

Konoha では、全てのオペレータはメソッドのシンタックスシュガーである。例えば、
加算演算 x + yは、次のように単純にメソッド opAdd()に置き換えられて実行されて

いる。

x.opAdd(y) // x + y

そこで、演算子と対応つけられた opで始まるメソッドを定義すれば、オペレータの振

る舞いを変更することも可能である。ただし、演算子の再定義は、Konohaではあまり積
極的に推奨されない。

x + y x.opAdd(y) x - y x.opSub(y)

x * y x.opMul(y) x / y x.opDiv(y)

x mod y x.opMod(y)

x++ x = x.opNext() x-- x = x.opPerv()

x == y x.opEq(y) x != y x.opNeq(y)

x < y x.opLt(y) x <= y x.opLte(y)

x > y x.opEq(y) x >= y x.opNeq(y)

x & y x.opLand(y) x | y x.opLor(y)

x ˆ y x.opXor(y) ˜x x.opLnot(y)

x << y x.opLshift(y) x >> y x.opRshift(y)

|x| x.getSize() y in? x x.opHas(y)

x[n] x.get(n) x[n] = y x.set(n)

x[] = y x.setAll(n) x[s..e] x.opSubsete(s,e)

x[s..<e] x.opSubset(s,e) x[s..+n] x.opOffset(s,n)

第 11章

クラス階層と抽象化

オブジェクト指向プログラミングの黎明期には、「クラス継承 (class inheritance)」は中
心的な関心事であった。しかし、オブジェクト指向プログラミングが普及し、多種多様な

オブジェクトのモデリング手法が実践される内、クラス継承はよくも悪くも「熱く語る」

ほどの話ではなくなった。

Konohaは、そんな時代に設計されたので、「クラス継承」に関して、必要最小限の機能
をサポートするのみにとどめている。（しかも、現在のところ、ほとんどクラス継承を使

う機会はない。）

第 12章

例外処理

例外は、動的な非局所ジャンプの一種である。

12.1 例外クラス Exception

Exceptionは、例外の種類、状態を表現するクラスである。
クラス Exceptionは、Konohaにおいて例外を実現する唯一のクラスである。例外の種
類は、例外クラス内部の表現として分類され、どの例外も Exceptionクラスのインスタン
スとなる。

>>> e = new Security!!();
>>> e.class
Exception
>>> e = new IO!!();
>>> e.class
Exception

12.2 throw文

Konohaは、プログラム実行中に何らかの異常を検出したとき、例外をスローする。例
えば、0除算が発生した場合、その異常状態は Arithmetic!!例外がスローされたこと

で知ることができる。

>>> a = 0;
>>> 1 / a;
** Arithmetic!!: Divide by Zero

同様に、プログラマ自身も、専用のステートメント throwを使うことによって、例外を
スローすることができる。最も標準的な例外のスローは、新しく例外クラスを作成し、そ

れを throw文に与える方法である。

>>> throw new Security!!("Something wrong");
** Security!!: Something wrong

Konohaの throw文では、文字列（エラーメッセージ）をそのまま投げることができる。
文字列の先頭が例外名のタグであれば、例外の種類が適切に判別され、そうでなければ単

純に Exception!!になる。

>>> throw "Security!!: Something wrong";
** Security!!: Something wrong

12.3 try-catch文

try-cacth文は、tryブロック内で発生した例外を捕捉（キャッチ）し、プログラムの正
常化を行うための制御構造である。

try {
// ブロックでスローされた例外に対する

}
catch(IO!! e) {

// プログラムの正常化
}

catch 節は、複数種類の例外をそれぞれ処理することができる。例外はソースコード
の順に先頭からマッチング処理を行われ、最初にマッチングした catch節が処理される。
マッチングは、e instanceof E!!による半順序マッチングであり、複数の catch 節
の条件にマッチする可能性があるが、最初のひとつ以外は catch節は無視される。

try {
// ブロックでスローされた例外に対する

}
catch(Security!! e) {

// プログラムの正常化
}
catch(SQL!! e) {

// プログラムの正常化
}
catch(IO!! e) {

// プログラムの正常化
}

また、どの catch節の例外処理にもマッチしなかった場合は、例外は捕捉されることな
く、そのまま継続してスローされ続ける。

12.4 finally節

第 13章

スクリプトと名前空間

スクリプトは、Konohaのプログラムの単位である。プログラムは、スクリプトファイ
ルに記述され、それがロードされたのち、コンパイルされたスクリプトは Scriptクラスと
名前空間 NameSpaceクラスによって管理される。本章では、Scriptと NameSpaceク
ラスを紹介しながら、Konoha特有のスクリプティング機能も紹介する。

13.1 名前空間 NameSpace

Konohaは、大規模なソフトウェア開発を想定しているわけでないが、シンプルかつ十
分な名前空間の機構を備えている。全てのスクリプトは、最初、デフォルトの名前空間

mainをもっている。

現在の名前空間は、システム変数__ns__で確認することができる。

>>> __ns__
"main"

システム変数__ns__は、NameSpace型で、現在の名前空間のオブジェクトを返す。
通常、名前空間は、using文などを通して自動的に一貫性をもって管理されている。

13.1.1 クラス識別子とニックネーム*

クラスは、class文で宣言されたとき、その名前空間を前置名にして宣言される。実は、
これが Konohaシステム内での正式なクラス識別子となる。

>>> class C ;
>>> C // 正式なクラス識別子
main.C

Konohaは、クラス識別子を型の名前として仕様できない。あくまでもクラス名を用い
る。それぞれの名前空間は、クラス名とクラス識別子を結びつけたニックネーム表をもっ

ている。

>>> class C
>>> %dump(__ns__) // ニックネーム表の確認
Script main.Script
C main.C

13.1.2 パッケージとクラス名*

パッケージは、パッケージ名を名前空間としてもったスクリプトである。例えば、math
パッケージ内のMathクラスは、クラス識別子は math.Mathとなる。

konohaパッケージ外の名前空間に属すクラス識別子は、自動的にニックネーム表に登
録されない。そこで、明示的にニックネーム表に登録する必要がある。

using文は、もしパッケージが未ロードであればパッケージのロードを行ったのち、指
定されたクラスをニックネーム表に登録するステートメントである。

>>> using math.Math;
>>> Math
math.Math
>>> %dump(__ns__)
Script main.Script
C main.C
Math math.Math

もし指定したパッケージ内にニックネーム表と同じクラス名が存在する場合は、既存の

名前を優先する。強制的に新しい名前を用いたい場合は、@Overrideを使うこともでき

る。ただし、予期せぬ混乱をまねく可能性があるのであまりおすすめできない。

もし、パッケージ内の全てのクラスをロードしたい場合は、ワイルドカードを用いる

ことができる。パッケージ内に多くのクラスが存在するときは、こちらを使うと便利で

ある。

>>> using math.*;

また、クラス名を省略すると、パッケージのみのロード、もしくは強制的な再ロードを

意味する。このときは、ニックネーム表は更新されない。

>>> using math;

13.2 スクリプトクラス Script

Konoha は、スクリプトファイルを読み始める前に、新しい Script クラスを生成す
る。このとき、名前空間ごとに異なるクラスを生成するため、名前空間が main であ

れば、クラス識別子は main.Script のように生成される。また、新しく生成された

main.Scriptは、Konohaが標準的に提供する基本パッケージ konoha.Scriptの性

質を継承している。

Script クラスとスクリプトファイルは、ちょうど class Script 宣言のブロック中

にスクリプトファイルが置かれたものと考えるとわかりやすい。グローバル（に見えた）

変数は、実は Scriptクラスのフィールド変数に相当し、関数も Scriptのメソッドとなる。
Konoha言語では、それぞれスクリプト変数、スクリプト関数と呼ばれる。

class Script

{
// スクリプトファイル の始まり

// スクリプトファイル の終わり
}

スクリプトファイルが、通常のクラス宣言と異なる点は、直接、ステートメントを記述

できる点である。（これは、スクリプトステートメントと呼ばれる。)例えば、次はスクリ
プトステートメントの例である。

for(i = 0; i < 10; i++) print i;

スクリプトステートメントは、無名関数によってラップされて実行される。つまり、上

のスクリプトステートメントは、次の lambda()関数定義と呼び出しと同義である。

Any lambda() {
for(i = 0; i < 10; i++) print i;

}
lambda();

注意：次節で述べるとおり、変数 iは、lambda()関数内のローカル変数となるため、

注意が必要である。

13.3 スクリプト変数

スクリプト変数は、Scriptクラスのフィールド変数である。グローバル変数のように振
る舞うが、実は名前空間が違えば、その名前空間に固有の Scriptオブジェクトを通してア

クセスしているため、参照されることはない。つまり、スクリプト変数のスコープは名前

空間によって分離されている。

// スクリプト変数
card = new int[54]; // スクリプト変数

Scriptクラスは、特別な@Singletonクラスとして導入されている。インスタンスの
数がひとつに限定されるため、実行中であっても一貫性を乱すことなくフィールド変数を

増やすことができる。その上限数は、おおよそ 128 変数くらいである。上限数がはっき
りしない理由は、Konohaの最適化オプションの状態で、UNBOXFIELDが採用された場
合、intと floatはそれぞれ 2変数分の領域を消費することがあるためである。どちらにし
ても、通常のプログラミングにおいてスクリプト変数を使い切ることはないし、そのよう

なスクリプトは（メンテナンス性がないため）書いてはいけない。

ちょっとしたテクニックであるが、スクリプト中で一次的に利用する変数は、ブロック

ステートメントなどを利用してローカル変数として扱うこともできる。また、スクリプト

変数が不足するようなプログラムはメンテナンス不能になるため、書いてはいけない。

card = new int[54]; // スクリプト変数
{

int i; 　　　　　　　　　　 // ローカル変数
for(i = 0; i < |card|; i++) {

card[i] = Int.random(i);
}

}

13.3.1 変数スコープ

Konohaは、単純化された３つの変数スコープをもつ。変数名は、まずローカル変数を
検索し、続いてフィールド変数、最後にスクリプト変数を検索する。明示的に、フィール

ド変数やスクリプト変数を参照したい場合は、変数名の前に_, __をそれぞれ付ける。も
しくは、フィールド変数、スクリプト変数を宣言するときにあらかじめ付けておいても構

わない。

>>> String s = "script"; // スクリプト変数
>>> class Person {
>>> String s = "field";　 // フィールド変数
... void test() {
... String s = "local"; // ローカ
ル変数
... print s, _s, __s;
... }
... }
>>> t = new T();
>>> t.test()

s="local", _s="field", __s="script"

13.3.2 定数の定義

スクリプトの重要な役割は、定数の定義である。Konoha では、名前ルールによって、
グローバル定数、ローカル定数、クラス定数がある。

GLOBAL_ =
LOCAL =
Math.PI_2 =

第 14章

デバッグ

スクリプティング言語は、一般に「プログラミングしやすい」言語とみなされている。

その理由のひとつは、ラピッドプロトタイプピングによる、よくテストしながらの開発ス

タイルにあると考えられる。しかし、どのようなソースコードにも必ずバグは存在し、モ

ダンなプログラミング言語は、バグを発見するのを支援する機能を備えている。

Konohaは、いくつかの、例によって、本格的なデバッグ機能ではないが、クラスルー
ム利用においてデバッグの基本技法を学ぶのに十分なデバッグ機能をサポートしている。

14.1 DEBUGブロックとデバッグモード

デバッグは、バグの原因を分析するデバッグ情報をえることから始まる。そのために

は、ソースコードを眺めているより、実際にプログラムを実行させて、内部状態の出力を

える方がバグを発見しやすい。

次は、簡単なバグの例である。プログラムが停止しなくなる理由を調べるため、引数 n

を表示している。

int fibo(int n) {
if(n < 0) print n; // デバッグ情報の出

力
if(n == 0) return 0;
return fibo(n-1) + fibo(n-2);

}

デバッグ情報は、特別な解析ツールを利用しない限り、プログラマ自身がプログラムし

て出力することになる。つまり、デバッグ情報の出力もソースコードの一部となる。こう

なると、正規のソフトウェア機能とデバッグ情報の出力パートが混在し、いろいろ不都合

が生じる。

Konoha では、DEBUG ブロックと呼ぶ専用のステートメントブロックを導入し、デ
バッグのためのソースコードを分離して書くことができる。これは、後から述べるとお

り、リリース/デバッグの実行モードを切り替えることで、デバッグ情報の出力を切り替え
ることが可能にする。

int fibo(int n) {
DEBUG { // デバッグ情報の出

力
if(n < 0) print n;

}
if(n == 0) return 0;
return fibo(n-1) + fibo(n-2);

}

14.1.1 デバッグモードとリリースモード

Konohaは、デバッグモードとリリースモードの 2種類の実行モードを持っている。デ
バッグモードは、開発者向けのモードであり、デバッグ情報を出力しながら実行する。リ

リースモードは、一般のユーザの利用を前提としたモードであり、逆にデバッグ情報を抑

制して実行するモードである。

Konohaは、通常、リリースモードでスクリプトを実行する。デバッグモードで実行す
る場合は、起動オプションに -gを用いる。

$ konoha -g script.k

注意：対話モードは、-gを付けなくてもデバッグモードとして起動される。

14.1.2 実行時のモード切り換え

Konoha内部では、デバッグ/リリースモードの切り換えは、実行コンテクスト (Context)
単位で行っている。そのため、Contextクラス関数を通して、実行中にモードを切り替え
ることもできる。

Context.setDebug(true); // デバッグモードに変更
Context.setDebug(false); // リリースモードに変更

「実行時のモード切り換え」を正しく使うためには、Konohaのスクリプト実行の内側ま
で注意する必要がある。スクリプトの実行は、厳密に言えば、コンパイル（コード生成）

とコードの実行からなり、リリース/デバッグモードはコンパイル時のコード生成にも適
用される。特に、リリースコンパイル時は、実行速度向上と利用メモリ量軽減を意図した

コード最適化のため、DEBUGブロックと（後述する）print文、assert文は無視される。
したがって、リリースコンパイルされたコードは、デバッグモードで実行してもデバッグ

情報は表示されない。

14.1.3 @Release アノテーション

クラス宣言やメソッド定義のとき、@Releaseアノテーションを付けると、その対象

となるクラスとメソッドは、デバッグコンパイルであっても常にリリース版のコードが

生成される。十分に開発が完了した部分は、@Releaseアノテーションを付けることで、

開発チームにもコンパイラにも開発の進捗を明確に伝えることができる。

@Release
class Person {

...
}

注意: クラスに、@Releaseアノテーションを付けると、そのクラスの全てのメソッド
がリリースとなる。

14.2 print 文

Konoha における正式なプリントアウト方法は、いわゆる OUT 定数を用いた Output-
Streamクラスを用いた標準出力である。一方、古式ゆかしい print文もサポートしてい
る。これは、クラスルームにおいて、「printデバッグ」を教えるとき print文がないと都合
がわるいからである。もちろん、誰にとっても「printデバッグ」は有効であり、Konoha
における print文は、「printデバッグ」専用の機能を備えている。

print文は正式なプログラムの一部ではなく、デバッグ/リリースモードによって、プリ
ントアウト出力のありなしが切り替わる。リリースコンパイルでは、そもそも print自体、
無視される。

もうひとつの特徴は、print文は、デバッグ情報として加工された情報が出力されるこ
とである。まず、必ず print 文が実行された位置が [ファイル名:行番号] として出力され
る。変数の値を出力するときは、（ご親切に）変数名も自動的に出力される。

>>> a = 0;
>>> print a;
[shell.c:4] a=0

単一の print文において、複数の情報を出力する場合は、カンマで区切る。変数名以外
の式を与えると、式の評価結果が表示される。

>>> a = 0;
>>> b = 1;
>>> print a, b, (a+b);
[shell.c:4] a=0, b=1, 1

14.2.1 @Date アノテーション*

print文は、@Dateアノテーションを付けることができる。@Dateアノテーションは、
print文出力に対し、デバッグ情報を出力した時刻を表示するオプションである。

>>> @Date print a, b, (a+b);
2009/12/24 00:00:00 [shell.c:4] a=0, b=1, 1

14.2.2 ロッギング**

Konoha プロジェクトでは、print 文とロギング (syslog) との統合を検討し、
@Log(ALART)のようなわかりやすい拡張オプションを提供する予定である。

>>> @Log(PANIC) print name;

14.3 assert文

アサーション (assertion) は、表明と訳される。assert 文は、プログラムが正しく動作
するための前提条件を表明し、プログラムが表明された条件通り正しく動作しているか確

認するデバッグ機能を提供する。

まず、正しいプログラムという概念を考えるため、簡単な例を考えてみたい。次の

fibo()関数は、引数 nにおいて n > 0を満たす整数が与えられるという前提でプログ

ラミングされている。しかし、利用者の方は原理的に fibo(-1)とコールすることもで

きる。（その場合は、StackOverflow!!がスローされる。）

>>> int fibo(int n) {
... if(n == 1 || n == 2) return 1;
... return fibo(n-1) + fibo(n-2);
... }
>>> fibo(-1)
** StackOverflow!!

ここで、「引数が負の数でも動作するように fibo()関数を作らなかった」方に責任が

あるのか、それとも「fibo(-1)をコールした」方が悪いのか、という責任問題が発生す

る。どちらにバグの責任があるかはっきりしない場合は、デバッグすることができない。

今回は、もし前者の立場をとれば、あらゆる引数値の範囲を確認し、その範囲を超えた

場合は、何らかの例外をスローする方法も考えられる。（ちなみに、勝手に引数の値を修

正してプログラムを継続する修正はあまり望ましいデバッグではない。）

int fibo(int n) {
if(n > 0) {

if(n == 1 || n == 2) return 1;
return fibo(n-1) + fibo(n-2);

}
throw new Arithmetic!!("Negative fibo");

}

逆に、負のフィボナッチ数列は通常、定義されていないため、そもそも fibo(-1)が

おかしいと考えてもよい。そのように考えても、プログラミング言語側はそこまで察する

ことはできない。そこで、何らかの方法で fibo(n)の正しい動作条件を表明しておく必

要がある。

assert文は、プログラムの正しい動作を表明するステートメントである。assert文に
続く、条件式には、正しい動作のとき trueになるように書く。

int fibo(int n) {
assert(n > 0);
if(n == 1 || n == 2) return 1;
return fibo(n-1) + fibo(n-2);

}

assert 文による表明は、実行時における動的な検査で判定される。もし表明に違反し
た場合は、Assertion!!例外がスローされ、プログラムの実行は正常化されるまで実行

が中断される。ただし、Assertion!!例外の標準的な対応は、表明に違反したコードを

探して、デバッグを行うことである。

>>> fibo(-1)
** Assertion!!: n > 0

注意すべき点は、assert文はデバッグ作業を助けるステートメントである点である。一
般に、assert文は、頻繁にコールされる関数に用いると、パフォーマンス低下の原因とな
る。そのため、リリースコンパイルでは除去してコンパイルされる。リリース前に、徹底

的にソフトウェアテストを行い、Assertion!!例外が発生しないようにデバッグする必

要がある。

14.3.1 リリース・アサーション

アサーションは、デバッグ用の機能である。しかし、リリース版であっても、致命的な

実行結果をもたらす前に、アサーション機能によって処理を中断させたいと考える場合も

ある。Konohaでは、@Releaseアノテーションを assert文の前に追加することで、リ
リースコンパイル時でも assert 文がコード生成されるようになる。

@Release assert(n > 0);

注意：頻繁にアクセスされる箇所に、リリース・アサーションを用いると、性能低下の

原因となる。リリース・アサーションの導入は慎重に検討すべきである。

14.3.2 assert 拡張文法

C/C++では、assert()は関数 (マクロ)で提供されている。Konohaでは、基本的に
互換性を保った形で、ただし条件式に続けてステートメントを記述できるように拡張して

ある。したがって、assert文の文法は次のとおりになる。
assert (expr) stmt

ここでは、まず条件式 expr を評価し、その結果が true なら何もおこらない。もし、
falseの場合、それに続くステートメント stmtが実行され、最後に Assertion!!例外

がスローされる。

何のための拡張文法かと言えば、要するに Assertion!!が発生したとき、その瞬間

のデバッグ情報を表示するためである。（ただし、意外と便利である。）

assert(n > MAX) {
print n, MAX;

}

14.4 utest文**

Konohaプロジェクトでは、ユニットテストをサポートするための専用のステートメン
トを導入する計画である。

@Name("test name")
utest (n > 0) { // テスト結果の表明

// テストコード
}

14.5 ブレークポイント**

Konohaの対話モードでは、デバッガ (gdb)の代表的なデバッグ機能をサポートする予
定である。これらの機能は、バイトコードの実行時書き換えの APIが整備したのち、拡張
される予定である。

Konohaライブラリ

Konoha は、C 言語のライブラリ (libkonoha, konoha.dll) として提供され、アプリケー
ションに組み込んで利用することができる。

15.1 Konohaインスタンス

Konoha インスタンスは、Konoha スクリプティング言語エンジンと実行コンテクス
トから構成され、konoha_t という型で定義されている。Konoha 言語の機能は、全て
Konohaインスタンスを通して実行される。
次は、最も簡単な Konohaインスタンスの利用例である。まず、Konohaインスタンス
の生成 (konoha_open()) し、Konoha ステートメントを実行 (konoha_eval()) し、
最後に Konohaインスタンスを解放 (konoha_close())している。

#include<konoha.h>
int main(void)
{

konoha_t k = konoha_open(0);
konoha_eval(k, "print ’hello,world’;");
konoha_close(k);
return 0;

}

Konohaインスタンスは、実行コンテクストを保持しているので、同じ Konohaインス
タンス上で定義されたスクリプト変数やスクリプト関数は、その定義を持続して評価が行

える。

#include<konoha.h>
int main(void)
{

konoha_t k = konoha_open(0);
konoha_eval(k, "a=1;");
konoha_eval(k, "b=2;");
konoha_eval(k, "print a + b;");
konoha_close(k);
return 0;

}

また、スクリプトファイルをロードする API も用意されている。スクリプトのロード
のみであるため、main()関数は実行されない。

#include<konoha.h>
int main(void)
{

konoha_t k = konoha_open(0);
konoha_include(k, "file.k");
konoha_close(k);
return 0;

}

15.1.1 マルチ・インスタンス

Konohaライブラリは、マルチ・インスタンス対応である。つまり、複数のインスタン
スを同時につくることが可能である。同時に作成できるインスタンスの数は、システムの

リソースが続く限り、無制限である。

#include<konoha.h>
int main(void)
{

konoha_t k1 = konoha_open(0);
konoha_t k2 = konoha_open(0);
...
konoha_close(k1);
konoha_close(k2);
return 0;

}

Konohaのインスタンスは、メモリ効率はあまりよくないが、完全に独立したメモリ領
域を持つように設計されている。したがって、マルチスレッド環境においては、スレッド

ごとに Konohaインスタンスを割り当てる限り、ブロッキングもレースコンディションも
起こらない。もし異なるスレッドから、同じ Konohaインスタンスを利用したときは、実

行コンテクストが混在して破滅的な結果を招くことがある。

15.2 Cからのスクリプト関数の利用*

Konohaで開発されたプログラムは、スクリプト関数を通して利用することができる。

konoha_fcall(k, "funcname(%s)", "value");
konoha_fcall_int(k, "funcname(%s)", "value");
konoha_fcall_float(k, "funcname(%s)", "value");
konoha_fcall_text(k, "funcname(%s)", "value");

C/C++ライブラリの利用

スクリプティング言語は、別名「グルー (glue)言語」とも呼ばれる。それは、C/C++な
ど、他のプログラミング言語で開発されたライブラリを「糊のごとく」はり合わせて利

用することが一般的な利用形態であるためである。実際、多くの C/C++ ライブラリが、
Konohaにバインドされて、Konohaから利用可能になっている。

16.1 C言語関数のバインド

最も簡単なバインドは、C言語の関数を Konohaのクラス関数へのバインドである。ま
ずは、C言語の標準ライブラリの math.hを例にしながら、バインド機構を紹介する。

16.1.1 パッケージの作成

Konoha では、バインドされた C/C++ ライブラリは、全てパッケージ機構で管理され
る。まず、パッケージ名を決めてパッケージスクリプトを開発する。

ここでは、mathパッケージとする。パッケージスクリプトのファイル名は、math.k

となる。新しく作成したパッケージは、次のどこかのディレクトリに保存する。

1. 環境変数
{$KONOHA_PACKAGE}/math/math.k

2. Konoha インストールディレクトリ
{$konoha.home}/package/math/math.k

3. ローカルパッケージ
˜/.konoha/math/math.k

4. 一時的なパッケージ
./.konoha/math/math.k

注意: パッケージスクリプトは、上の順番で検索され、最初に発見したスクリプトが
ロードされる。今回の例としている math パッケージは、インストールされているため、

実際に試している読者はパッケージ名を変更するなど工夫が必要である。

さて、肝心のパッケージスクリプト math.kの内容であるが、特別なことはない。単

純に、バインドする (予定の)関数を Konohaからみた名前や型で設計するだけである。メ
ソッドの実体は、バインドするため、抽象メソッドとして宣言するのみである。

class Math;
@Static float Math.sin(float x);
@Static float Math.abs(float x);
...

注意：Konohaは、オブジェクト指向プログラミング言語であるため、クラスが機能の
単位の中心となる。そのため、非オブジェクト指向言語の C関数であっても、何らかのク
ラス関数としてバインドする必要がある。（スクリプト関数にバインドすることはできな

い。）上記の例では、Mathクラスをクラス関数のバインド先として定義している。

16.1.2 グルー関数の作成

Konoha の内部では、メソッドは knh_fmethod 型で定義された関数で実装されてい

る。次は、konoha.h ヘッダファイルに定義されている knh_fmethod 型の抜粋である。

コンテクストとスタックフレームポインタを引数として受け取り、戻り値は void型であ
る。（ただし、一読してメソッドとわかるためと fastcallの切り換えのため、METHODマ
クロで定義されている。）

#define METHOD void KNH_CC_FASTCALL
typedef METHOD (*knh_fmethod)(Ctx *, knh_sfp_t *);

グルー関数とは、knh_fmethod 型に準拠したラッパー関数のことである。その機能

は、単に、Konoha側のスタックフレームポインタで渡されたパラメータを C言語の型に
変換し、C関数を実行し、そして C関数の実行結果を再び、Konoha側のスタックに戻し
ている。

// Static float Math.abs(float x);
METHOD Math_abs(Ctx *ctx, knh_sfp_t *sfp)
{

double v = (double)sfp[1].fvalue; // C言語の型へ
変換

v = fabs(v);
KNH_RETURN_Float(ctx, sfp, v); // Konohaのス

タックへ戻す
}

グルー関数の仕様は、次節で詳しく述べるが、sfp[1]には第一引数、sfp[2]には第

二引数、sfp[n]には第 n引数の値が格納されている。例えば、次のような場合、第一引数
の値は、-1.0であり、それは sfp[1].fvalueに格納されている。

>>> Math.abs(-1.0)
1.0000

グルー関数の特徴は、型チェック済みの引数が必ず渡される点である。これは、パッ

ケージスクリプトで宣言したメソッド定義に基づいている。つまり、グルー関数の開発者

は、特別な場合をのぞいて sfp[1]の型をチェックする必要はない。

逆に、戻り値は、グルー関数の開発者が型チェック済みの値を正しく返さなけ

ればならない。Math.abs() の戻り値の型は、float であるため、専用のマクロ
KNH_RETURN_Float を用いて、Konoha スタックに戻り値を設定し、同時にグルー関
数から抜け出している。

注意：グルー関数内で設定する戻り値の型が間違っていた場合は、ほぼ例外なくクラッ

シュする原因となる。グルー関数の作者は細心の注意を払う必要がある。

16.1.3 共有ライブラリのコンパイル

16.2 グルー関数とメソッド

16.3 構造体とクラス

16.4 関数ポインタとクロージャ

付録 A

入手方法

$ tar zxfv konoha-0.x.y.tar.gz

$ cd konoha-0.x.y

$ ./configure

$ make

$ sudo make install

